

# Natural

## Systemfunktionen

Version 9.1.2

April 2023

Dieses Dokument gilt für Natural ab Version 9.1.2.

Hierin enthaltene Beschreibungen unterliegen Änderungen und Ergänzungen, die in nachfolgenden Release Notes oder Neuauflagen bekanntgegeben werden.

Copyright © 1979-2023 Software AG, Darmstadt, Deutschland und/oder Software AG USA, Inc., Reston, VA, USA, und/oder ihre Tochtergesellschaften und/oder ihre Lizenzgeber.

Der Name Software AG und die Namen der Software AG Produkte sind Marken der Software AG und/oder Software AG USA Inc., einer ihrer Tochtergesellschaften oder ihrer Lizenzgeber. Namen anderer Gesellschaften oder Produkte können Marken ihrer jeweiligen Schutzrechtsinhaber sein.

Nähere Informationen zu den Patenten und Marken der Software AG und ihrer Tochtergesellschaften befinden sich unter <http://documentation.softwareag.com/legal/>.

Diese Software kann Teile von Software-Produkten Dritter enthalten. Urheberrechtshinweise, Lizenzbestimmungen sowie zusätzliche Rechte und Einschränkungen dieser Drittprodukte können dem Abschnitt "License Texts, Copyright Notices and Disclaimers of Third Party Products" entnommen werden. Diese Dokumente enthalten den von den betreffenden Lizenzgebern oder den Lizenzen wörtlich vorgegebenen Wortlaut und werden daher in der jeweiligen Ursprungssprache wiedergegeben. Für einzelne, spezifische Lizenzbeschränkungen von Drittprodukten siehe PART E der Legal Notices, abrufbar unter dem Abschnitt "License Terms and Conditions for Use of Software AG Products / Copyrights and Trademark Notices of Software AG Products". Diese Dokumente sind Teil der Produktdokumentation, die unter <http://softwareag.com/licenses> oder im Verzeichnis der lizenzierten Produkte zu finden ist.

Die Nutzung dieser Software unterliegt den Lizenzbedingungen der Software AG. Diese Bedingungen sind Bestandteil der Produktdokumentation und befinden sich unter <http://softwareag.com/licenses> und/oder im Wurzelverzeichnis des lizenzierten Produkts.

**Dokument-ID: NATMF-NNATFUNCTIONS-912-20230424DE**

# Inhaltsverzeichnis

|  |    |
|--|----|
| Vorwort .....  | v  |
| 1 Über diese Dokumentation .....   | 1  |
| Dokumentationskonventionen .....   | 2  |
| Online-Informationen und Support .....                                     | 2  |
| Datenschutz .....  | 3  |
| I .....  | 5  |
| 2 Systemfunktionen für Verarbeitungsschleifen .....                        | 7  |
| Systemfunktionen für Verarbeitungsschleifen benutzen .....                 | 8  |
| AVER(r)(field) .....   | 10 |
| COUNT(r)(field) .....  | 10 |
| MAX(r)(field) .....  | 10 |
| MIN(r)(field) .....  | 11 |
| NAVER(r)(field) .....  | 11 |
| NCOUNT(r)(field) .....   | 11 |
| NMIN(r)(field) .....   | 12 |
| OLD(r)(field) .....  | 12 |
| SUM(r)(field) .....  | 12 |
| TOTAL(r)(field) .....  | 13 |
| Beispiele .....  | 13 |
| 3 Mathematische Systemfunktionen .....                                     | 19 |
| II Verschiedene Systemfunktionen .....                                     | 23 |
| 4 *MINVAL/*MAXVAL - Minimal-/Maximalwertes eines Feldes .....              | 25 |
| Funktion .....   | 26 |
| Einschränkungen .....  | 26 |
| Syntax-Beschreibung .....  | 26 |
| Tabelle für die Konvertierung der resulting-format/length-Werte .....      | 28 |
| Berechnung der result-format-length .....                                  | 30 |
| 5 *TRANSLATE - Umsetzung in Groß-/Kleinschreibung .....                    | 35 |
| Funktion .....   | 36 |
| Einschränkungen .....  | 36 |
| Syntax-Beschreibung .....  | 36 |
| Beispiel .....   | 37 |
| 6 *TRIM - Entfernen von führenden und/oder nachfolgenden Leerstellen ..... | 39 |
| Funktion .....   | 40 |
| Einschränkungen .....  | 40 |
| Syntax-Beschreibung .....  | 40 |
| Beispiel .....   | 41 |
| 7 POS - Feldidentifikationsfunktion .....                                  | 45 |
| 8 RET - Returncode-Funktion .....  | 47 |
| 9 SORTKEY - Sort-Key Function .....  | 49 |
| III Funktionen, die als Natural-Objekte ausgeliefert werden .....          | 53 |
| 10 Funktionen, die als Natural-Objekte ausgeliefert werden .....           | 55 |
| URL-Kodierung .....  | 56 |

|                            |    |
|----------------------------|----|
| Base64-Kodierung .....     | 67 |
| Stichwortverzeichnis ..... | 75 |

---

# Vorwort

---

Diese Dokumentation beschreibt in Natural eingebaute Funktionen, die in bestimmten Statements benutzt werden können, siehe *Systemfunktionen* im *Leitfaden zur Programmierung*.

Diese Dokumentation ist in die folgenden Abschnitte untergliedert:

**Systemfunktionen für Verarbeitungsschleifen**

Beschreibt Systemfunktionen, die im Zusammenhang mit einer Programmschleife verwendet werden können.

**Mathematische Systemfunktionen**

Beschreibt Systemfunktionen, die in arithmetischen Statements und in logischen Bedingungen (Logical Condition Criteria, LCC) unterstützt werden.

**Verschiedene Systemfunktionen**

Beschreibt Systemfunktionen zur Auswertung des Minimums/Maximums eines Feldes; Systemfunktion zur Feldidentifizierung; Systemfunktion zum Empfangen des Return Codes von einem Nicht-Natural-Programm; Systemfunktion zum Konvertieren von „nicht richtig sortierten“ Zeichen in andere Zeichen, die vom Sortierprogramm oder Datenbanksystem alphabetisch „richtig sortiert“ werden; Systemfunktion zur Umsetzung von Klein- auf Großbuchstaben; Systemfunktion zum Entfernen von führenden und/oder nachfolgenden Leerzeichen.

**Funktionen, die als Natural-Objekte ausgeliefert werden**

Beschreibt Funktionen, die als Natural-Objekte ausgeliefert werden, um zum Beispiel die URL-Kodierung und die Base64-Konvertierung zu unterstützen.

Siehe auch *Beispiel für Systemvariablen und Systemfunktionen* im *Leitfaden zur Programmierung*:

---

# 1 Über diese Dokumentation

---

- Dokumentationskonventionen ..... 2
- Online-Informationen und Support ..... 2
- Datenschutz ..... 3

## Dokumentationskonventionen

---

| Konvention                 | Beschreibung  |
|----------------------------|---|
| <b>Fettschrift</b>         | >Kennzeichnet Elemente auf einem Bildschirm.  |
| Nichtproportionale Schrift | Kennzeichnet Namen und Orte von Diensten im Format <i>Ordner.Unterordner.Dienst</i> , Programmierschnittstellen (APIs), Namen von Klassen, Methoden und Properties in Java.   |
| <i>Kursivschrift</i>       | Kennzeichnet:<br><br>Variablen, für die Sie situations- oder umgebungsspezifische Werte angeben müssen.<br>Neue Begriffe, wenn sie erstmals im Text auftreten.<br>Verweise auf andere Dokumentationsquellen.        |
| Nichtproportionale Schrift | Kennzeichnet:<br><br>Text, den Sie eingeben müssen.<br>Meldungen, die vom System angezeigt werden.<br>Programmcode.   |
| { }                        | Zeigt eine Reihe von Auswahlmöglichkeiten an, von denen Sie eine auswählen müssen. Geben Sie nur die innerhalb der geschweiften Klammern vorhandenen Informationen ein. Geben Sie nicht die Klammersymbole { } ein. |
|                            | Trennt zwei sich gegenseitig ausschließende Auswahlmöglichkeiten in einer Syntaxzeile voneinander ab. Geben Sie eine der Auswahlmöglichkeiten ein. Geben Sie nicht das Symbol   ein.                                |
| [ ]                        | Zeigt eine oder mehrere Optionen an. Geben Sie nur die innerhalb der eckigen Klammern vorhandenen Informationen ein. Geben Sie nicht die Klammersymbole [ ] ein.  |
| ...                        | Zeigt an, dass Sie mehrere Auswahlmöglichkeiten desselben Typs eingeben können. Geben Sie nur die Informationen ein. Geben Sie nicht die drei Auslassungspunkte (...) ein.  |

## Online-Informationen und Support

---

### Produktdokumentation

Sie finden die Produktdokumentation auf unserer Dokumentationswebsite unter <https://documentation.softwareag.com>.

Zusätzlich können Sie auch über <https://www.softwareag.cloud> auf die Dokumentation für die Cloud-Produkte zugreifen. Navigieren Sie zum gewünschten Produkt und gehen Sie dann, je nach Produkt, zu „Developer Center“, „User Center“ oder „Documentation“.



## Produktschulungen

Sie finden hilfreiches Produktschulungsmaterial auf unserem Lernportal unter <https://knowledge.softwareag.com>.

## Tech Community

Auf der Website unserer Tech Community unter <https://techcommunity.softwareag.com> können Sie mit Experten der Software AG zusammenarbeiten. Von hier aus können Sie zum Beispiel:

- Unsere umfangreiche Wissensdatenbank durchsuchen.
- In unseren Diskussionsforen Fragen stellen und Antworten finden.
- Die neuesten Nachrichten und Ankündigungen der Software AG lesen.
- Unsere Communities erkunden.
- Unsere öffentlichen Repositories auf GitHub and Docker unter <https://github.com/softwareag> und <https://hub.docker.com/publishers/softwareag> besuchen und weitere Ressourcen der Software AG entdecken.

## Produktsupport

Support für die Produkte der Software AG steht lizenzierten Kunden über unser Empower-Portal unter <https://empower.softwareag.com> zur Verfügung. Für viele Dienstleistungen auf diesem Portal benötigen Sie ein Konto. Wenn Sie noch keines haben, dann können Sie es unter <https://empower.softwareag.com/register> beantragen. Sobald Sie ein Konto haben, können Sie zum Beispiel:

- Produkte, Aktualisierungen und Programmkorrekturen herunterladen.
- Das Knowledge Center nach technischen Informationen und Tipps durchsuchen.
- Frühwarnungen und kritische Alarmer abonnieren.
- Supportfälle öffnen und aktualisieren.
- Anfragen für neue Produktmerkmale einreichen.

## Datenschutz

---

Die Produkte der Software AG stellen Funktionen zur Verarbeitung von personenbezogenen Daten gemäß der Datenschutz-Grundverordnung (DSGVO) der Europäischen Union zur Verfügung. Gegebenenfalls sind in der betreffenden Systemverwaltungsdokumentation entsprechende Schritte dokumentiert.

---

# I

---

|   |    |
|---|----|
| ■ 2 Systemfunktionen für Verarbeitungsschleifen ..... | 7  |
| ■ 3 Mathematische Systemfunktionen .....              | 19 |

---

## 2 Systemfunktionen für Verarbeitungsschleifen

---

|  |    |
|--|----|
| ▪ Systemfunktionen für Verarbeitungsschleifen benutzen ..... | 8  |
| ▪ AVER(r)(field) .....                                       | 10 |
| ▪ COUNT(r)(field) .....                                      | 10 |
| ▪ MAX(r)(field) .....  | 10 |
| ▪ MIN(r)(field) .....  | 11 |
| ▪ NAVER(r)(field) .....                                      | 11 |
| ▪ NCOUNT(r)(field) .....                                     | 11 |
| ▪ NMIN(r)(field) .....                                       | 12 |
| ▪ OLD(r)(field) .....  | 12 |
| ▪ SUM(r)(field) .....  | 12 |
| ▪ TOTAL(r)(field) .....                                      | 13 |
| ▪ Beispiele .....  | 13 |

Dieses Kapitel erläutert die Natural-Systemfunktionen, die im Zusammenhang mit einer Programmschleife verwendet werden können.

## Systemfunktionen für Verarbeitungsschleifen benutzen

---

- Spezifikation/Auswertung
- Systemfunktionen im SORT GIVE-Statement
- Arithmetischer Überlauf bei AVER, NAVER, SUM oder TOTAL
- Statement-Referenzierung (r)

### Spezifikation/Auswertung

Natural-Systemfunktionen können angegeben werden in

- zuweisenden und arithmetischen Statements:

- MOVE
- ASSIGN
- COMPUTE
- ADD
- SUBTRACT
- MULTIPLY
- DIVIDE

- in Eingabe-Ausgabe-Statements:

- DISPLAY
- PRINT
- WRITE

die in einem der folgenden Statement-Blöcke stehen:

- AT BREAK
- AT END OF DATA
- AT END OF PAGE

d.h. für alle Verarbeitungsschleifen in den Statements FIND, READ, HISTOGRAM, SORT oder READ WORK FILE.

Wenn eine Systemfunktion in einem AT END OF PAGE-Statement verwendet wird, muss das betreffende DISPLAY-Statement eine GIVE SYSTEM FUNCTIONS-Klausel enthalten.

Datensätze, die aufgrund einer `WHERE`-Klausel zurückgewiesen werden, werden von einer Systemfunktion nicht ausgewertet.

Wenn mittels Systemfunktionen Datenbankfelder ausgewertet werden, die aus `FIND`-, `READ`-, `HISTOGRAM`- oder `SORT`-Schleifen stammen, die auf verschiedenen Ebenen liegen, werden die Feldwerte jeweils entsprechend ihrer Position in der Verarbeitungsschleifen-Hierarchie verarbeitet. Werte einer äußeren Schleife werden zum Beispiel nur dann verarbeitet, wenn für diese Schleife neue Datenwerte erhalten werden.

Wird eine Benutzervariable vor der ersten Verarbeitungsschleife definiert, wird sie für Systemfunktionen in der Schleife ausgewertet, in der das `AT BREAK`-, `AT END OF DATA`- oder `AT END OF PAGE`-Statement steht; wird eine Benutzervariable innerhalb einer Schleife definiert, wird sie in der gleichen Weise wie ein Datenbankfeld in der derselben Schleife behandelt.

Bei dem selektiven Einsatz von Systemfunktionen zur Auswertung von Benutzervariablen empfiehlt es sich, eine bestimmte Verarbeitungsschleife zu referenzieren (mittels Statement-Label oder Quellcode-Zeilenummer), um genau festzulegen, in welcher Schleife der Wert der Benutzervariablen ausgewertet werden soll.

### Systemfunktionen im `SORT GIVE`-Statement

Eine Systemfunktion kann auch referenziert werden, nachdem sie in der `GIVE`-Klausel eines `SORT`-Statements ausgewertet wurde.

In diesem Fall muss dem Namen der Systemfunktion bei der Referenzierung ein Stern (\*) vorangestellt werden.

### Arithmetischer Überlauf bei `AVER`, `NAVER`, `SUM` oder `TOTAL`

Ein Feld, das Sie mit den Systemfunktionen `AVER`, `NAVER`, `SUM` oder `TOTAL` verwenden, muss lang genug sein (standardmäßig oder vom Benutzer definiert), um die Summe der Feldwerte aufzunehmen. Falls der addierte Wert die Länge des Feldes überschreitet, erhalten Sie eine Fehlermeldung.

Normalerweise hat die Systemfunktion dieselbe Länge wie das angegebene Feld; ist diese Länge nicht ausreichend, dann verwenden Sie den Parameter `NL` des `SORT GIVE`-Statements, um die Ausgabelänge zu vergrößern:

```
SUM(field)(NL=nn)
```

Dadurch vergrößert sich nicht nur die Ausgabelänge, sondern auch die interne Länge des Feldes.

## Statement-Referenzierung (r)

Statement-Referenzierung kann auch auf bei Systemfunktionen angewendet werden. Weitere Einzelheiten entnehmen Sie dem Unterabschnitt *Datenbankfelder mit der (r)-Notation referenzieren - Notation (r)* im Abschnitt *Benutzervariablen im Natural Leitfadens zur Programmierung*.

Durch Verwendung eines Statement-Labels oder Angabe der Quellcode-Zeilenummer (r) können Sie bestimmen, in welcher Verarbeitungsschleife die jeweilige Systemfunktion für das betreffende Feld ausgewertet werden soll.

## AVER(r)(field)

---

|               |   |
|---------------|---|
| Format/Länge: | Wie Feld.<br><br>Ausnahme: bei einem Feld mit Format N hat $AVER(field)$ Format P (mit derselben Länge wie das Feld). |
|---------------|---|

Diese Systemfunktion enthält den Durchschnittswert (Average) aller Werte des angegebenen Feldes. AVER wird aktualisiert, wenn die Bedingung, unter der AVER angefordert wurde, erfüllt ist.

## COUNT(r)(field)

---

|               |    |
|---------------|----|
| Format/Länge: | P7 |
|---------------|----|

Diese Systemfunktion zählt die Durchläufe einer Verarbeitungsschleife. Der Wert von COUNT erhöht sich jedesmal um 1, wenn die Verarbeitungsschleife, in der sich COUNT befindet, durchlaufen wird, und zwar unabhängig vom Wert des mit COUNT angegebenen Feldes.

## MAX(r)(field)

---

|               |           |
|---------------|-----------|
| Format/Länge: | Wie Feld. |
|---------------|-----------|

Diese Systemfunktion enthält den größten Wert des angegebenen Feldes. MAX wird (falls erforderlich) jedesmal aktualisiert, wenn die Verarbeitungsschleife, in der sich MAX befindet, ausgeführt wird.



## MIN(r)(field)

---

|               |           |
|---------------|-----------|
| Format/Länge: | Wie Feld. |
|---------------|-----------|

Diese Systemfunktion enthält den kleinsten Wert des angegebenen Feldes. MIN wird (falls erforderlich) jedesmal aktualisiert, wenn die Verarbeitungsschleife, in der sich MIN befindet, ausgeführt wird.

## NAVER(r)(field)

---

|               |   |
|---------------|---|
| Format/Länge: | Wie Feld.<br><br>Ausnahme: bei einem Feld mit Format N hat NAVER( <i>field</i> ) Format P (mit derselben Länge wie das Feld). |
|---------------|---|

Diese Systemfunktion enthält den Durchschnittswert (Average) aller Werte des angegebenen Feldes, wobei Nullwerte nicht berücksichtigt werden. NAVER wird aktualisiert, wenn die Bedingung, unter der NAVER angefordert wurde, erfüllt ist.

## NCOUNT(r)(field)

---

|               |    |
|---------------|----|
| Format/Länge: | P7 |
|---------------|----|

Diese Systemfunktion zählt die Durchläufe einer Verarbeitungsschleife. Der Wert von NCOUNT erhöht sich jedesmal um 1, wenn die Verarbeitungsschleife, in der sich NCOUNT befindet, durchlaufen wird, wobei Durchläufe, bei denen der Wert des angegebenen Feldes Null ist, nicht mitgezählt werden.

Ob das Ergebnis von NCOUNT ein Array oder ein Skalarwert ist, hängt vom Argument (*field*) ab. Die Anzahl der resultierenden Ausprägungen ist dieselbe wie bei Feld.

## NMIN(r)(field)

---

|               |           |
|---------------|-----------|
| Format/Länge: | Wie Feld. |
|---------------|-----------|

Diese Systemfunktion enthält den kleinsten Wert des angegebenen Feldes, wobei Nullwerte nicht berücksichtigt werden. NMIN wird (falls erforderlich) jedesmal aktualisiert, wenn die Verarbeitungsschleife, in der sich NMIN befindet, ausgeführt wird.

## OLD(r)(field)

---

|               |           |
|---------------|-----------|
| Format/Länge: | Wie Feld. |
|---------------|-----------|

Diese Systemfunktion enthält den „alten“ Wert des angegebenen Feldes, d.h. den Wert, den das Feld vor einem in einer AT BREAK-Bedingung spezifizierten Gruppenwechsel (Wechsel des Feldwertes) bzw. vor einer Seitenende- oder Datenende-Bedingung (END OF PAGE, END OF DATA) hatte.

## SUM(r)(field)

---

|               |   |
|---------------|---|
| Format/Länge: | Wie Feld.   |
|               | Ausnahme: bei einem Feld mit Format N hat $SUM(field)$ Format P (mit derselben Länge wie das Feld). |

Diese Systemfunktion enthält die Summe aller Werte des angegebenen Feldes. SUM wird jedesmal aktualisiert, wenn die Verarbeitungsschleife, in der sich SUM befindet, ausgeführt wird. SUM wird nach jedem AT BREAK-Gruppenwechsel wieder auf Null gesetzt, addiert also nur Werte zwischen zwei Gruppenwechseln.



```

WRITE 22T 'TOTAL (ALL RECORDS):'
      T*SALARY TOTAL(SALARY(1))  CURR-CODE(1)
END-ENDDATA
END-READ
*
END      ↵

```

**Ausgabe des Programms ATBEX3:**

```

CITY                NAME                SALARY  CURRENCY
-----
SALT LAKE CITY      ANDERSON                50000  USD
SALT LAKE CITY      SAMUELSON               24000  USD

S A L T   L A K E   C I T Y      MINIMUM:    24000  USD
                                  AVERAGE:    37000  USD
                                  MAXIMUM:    50000  USD
                                  SUM:        74000  USD
                                  2 RECORDS FOUND

SAN DIEGO           GEE                     60000  USD

S A N   D I E G O           MINIMUM:    60000  USD
                                  AVERAGE:    60000  USD
                                  MAXIMUM:    60000  USD
                                  SUM:        60000  USD
                                  1 RECORDS FOUND

TOTAL (ALL RECORDS):    134000  USD      ↵

```

**Beispiel 2 – AT BREAK-Statement mit Natural-Systemfunktion AVER**

```

** Example 'ATBEX4': AT BREAK (with Natural system functions)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CITY
  2 SALARY (2)
*
1 #INC-SALARY (P11)
END-DEFINE
*
LIMIT 4
EMPL. READ EMPLOY-VIEW BY CITY STARTING FROM 'ALBU'
  COMPUTE #INC-SALARY = SALARY (1) + SALARY (2)
  DISPLAY NAME CITY SALARY (1:2) 'CUMULATIVE' #INC-SALARY
  SKIP 1
  /*
AT BREAK CITY

```

```

WRITE NOTITLE
  'AVERAGE:'          T*SALARY (1)  AVER(SALARY(1)) /
  'AVERAGE CUMULATIVE:' T*#INC-SALARY AVER(EMPL.) (#INC-SALARY)
END-BREAK
END-READ
*
END

```

#### Ausgabe des Programms ATBEX4:

| NAME                | CITY        | ANNUAL         | CUMULATIVE<br>SALARY |
|---------------------|-------------|----------------|----------------------|
| HAMMOND             | ALBUQUERQUE | 22000<br>20200 | 42200                |
| ROLLING             | ALBUQUERQUE | 34000<br>31200 | 65200                |
| FREEMAN             | ALBUQUERQUE | 34000<br>31200 | 65200                |
| LINCOLN             | ALBUQUERQUE | 41000<br>37700 | 78700                |
| AVERAGE:            |             | 32750          |                      |
| AVERAGE CUMULATIVE: |             |                | 62825                |

#### Beispiel 3 – AT END OF DATA-Statement mit Systemfunktionen MAX, MIN, AVER

```

** Example 'AEDEXIS': AT END OF DATA
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
  2 SALARY (1)
  2 CURR-CODE (1)
END-DEFINE
*
LIMIT 5
EMP. FIND EMPLOY-VIEW WITH CITY = 'STUTTGART'
  IF NO RECORDS FOUND
    ENTER
  END-NOREC
  DISPLAY PERSONNEL-ID NAME FIRST-NAME
    SALARY (1) CURR-CODE (1)
/*
AT END OF DATA

```

```

IF *COUNTER (EMP.) = 0
  WRITE 'NO RECORDS FOUND'
  ESCAPE BOTTOM
END-IF
WRITE NOTITLE / 'SALARY STATISTICS:'
              / 7X 'MAXIMUM:' MAX(SALARY(1)) CURR-CODE (1)
              / 7X 'MINIMUM:' MIN(SALARY(1)) CURR-CODE (1)
              / 7X 'AVERAGE:' AVER(SALARY(1)) CURR-CODE (1)

END-ENDDATA
/*
END-FIND
*
END

```

**Ausgabe des Programms AEDEX1S:**

| PERSONNEL ID       | NAME     | FIRST-NAME | ANNUAL SALARY | CURRENCY CODE |
|--------------------|----------|------------|---------------|---------------|
| 11100328           | BERGHAUS | ROSE       | 70800         | DM            |
| 11100329           | BARTHEL  | PETER      | 42000         | DM            |
| 11300313           | AECKERLE | SUSANNE    | 55200         | DM            |
| 11300316           | KANTE    | GABRIELE   | 61200         | DM            |
| 11500304           | KLUGE    | ELKE       | 49200         | DM            |
| SALARY STATISTICS: |          |            |               |               |
|                    | MAXIMUM: | 70800      | DM            |               |
|                    | MINIMUM: | 42000      | DM            |               |
|                    | AVERAGE: | 55680      | DM            |               |

**Beispiel 4 – AT END OF PAGE-Statement mit Systemfunktion AVER**

```

** Example 'AEPEX1S': AT END OF PAGE (structured mode)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 JOB-TITLE
  2 SALARY (1)
  2 CURR-CODE (1)
END-DEFINE
*
FORMAT PS=10
LIMIT 10
READ EMPLOY-VIEW BY PERSONNEL-ID FROM '20017000'
  DISPLAY NOTITLE GIVE SYSTEM FUNCTIONS
    NAME JOB-TITLE 'SALARY' SALARY(1) CURR-CODE (1)
/*
AT END OF PAGE

```

```
WRITE / 28T 'AVERAGE SALARY: ...' AVER(SALARY(1)) CURR-CODE (1)
END-ENDPAGE
END-READ
*
END      ←
```

**Ausgabe des Programms AEPEX1S:**

| NAME    | CURRENT<br>POSITION | SALARY | CURRENCY  | CODE |
|---------|---------------------|--------|-----------|------|
| CREMER  | ANALYST             |        | 34000 USD |      |
| MARKUSH | TRAINEE             |        | 22000 USD |      |
| GEE     | MANAGER             |        | 39500 USD |      |
| KUNEY   | DBA                 |        | 40200 USD |      |
| NEEDHAM | PROGRAMMER          |        | 32500 USD |      |
| JACKSON | PROGRAMMER          |        | 33000 USD |      |
|         | AVERAGE SALARY: ... |        | 33533 USD | ←    |





# 3 Mathematische Systemfunktionen

In logischen Bedingungen und den Arithmetik-Statements ADD, COMPUTE, DIVIDE, MULTIPLY und SUBTRACT können Sie die folgenden mathematischen Funktionen verwenden:

| Funktion             | Format/Länge                  | Ausgegebener Wert   |
|----------------------|-------------------------------|---|
| ABS( <i>field</i> )  | wie Feld ( <i>field</i> )     | Absoluter Wert eines Feldes.  |
| ATN( <i>field</i> )  | F8                            | Arcustangens eines Feldes.  |
| COS( <i>field</i> )  | F8                            | Kosinus eines Feldes.<br>Ist der Wert des Feldes größer oder gleich $10^{17}$ , ist COS( <i>field</i> ) "1".  |
| EXP( <i>field</i> )  | F8                            | Potenz eines Feldes ( <i>e field</i> ) mit der Basis <i>e</i> , wobei <i>e</i> die Basis natürlicher Logarithmen ist.   |
| FRAC( <i>field</i> ) | wie Feld ( <i>field</i> )     | Bruchteil (hinter dem Komma) eines Feldes.  |
| INT( <i>field</i> )  | wie Feld ( <i>field</i> )     | Ganzzahliger Teil eines Feldes (Integer).   |
| LOG( <i>field</i> )  | F8                            | Natürlicher Logarithmus eines Feldes.<br>Ein negativer Wert im Argumentfeld wird als positiver Wert behandelt.  |
| SGN( <i>field</i> )  | wie Feld ( <i>field</i> )     | Vorzeichen (Sign) eines Feldes (-1, 0, +1).   |
| SIN( <i>field</i> )  | F8                            | Sinus eines Feldes.<br>Ist der Wert des Feldes größer oder gleich $10^{17}$ , ist SIN( <i>field</i> ) "0".  |
| SQRT( <i>field</i> ) | (*)                           | Quadratwurzel eines Feldes (Square Root).<br>Ein negativer Feldwert wird wie ein positiver behandelt.   |
| TAN( <i>field</i> )  | F8                            | Tangens eines Feldes.<br>Ist der Wert des Feldes größer oder gleich $10^{17}$ , ist TAN( <i>field</i> ) "0".  |
| VAL( <i>field</i> )  | wie Zielfeld ( <i>field</i> ) | Numerischer Wert eines alphanumerischen Feldes. Der Wert des Feldes muss ein in alphanumerischer Form (Codepage oder Unicode) dargestellter numerischer Wert sein. Leerstellen vor und nach dem |

| Funktion | Format/Länge | Ausgegebener Wert  |
|----------|--------------|--|
|          |              | <p>Komma im Feld werden ignoriert. Komma (Dezimalpunkt) und Vorzeichen werden mitverarbeitet.</p> <p>Wenn das Zielfeld nicht lang genug ist, werden Nachkommastellen abgeschnitten (vgl. <i>Abschneiden und Runden von Feldwerten</i> im Abschnitt <i>Regeln für arithmetische Operationen im Leitfaden zur Programmierung</i>).</p> |

\*Diese Funktion wird wie folgt ausgewertet:

- Wenn das Feld Format/Länge F4 hat, hat  $SQRT(field)$  auch Format/Länge F4.
- Wenn das Feld Format/Länge F8 oder I hat, hat  $SQRT(field)$  Format/Länge F8.
- Wenn das Feld Format N oder P hat, siehe Informationen zu Format/Länge von  $SQRT(field)$  im Abschnitt *Genauigkeit der Ergebnisse bei arithmetischen Operationen* im Leitfaden zur Programmierung.

hat  $SQRT(field)$  Format/Länge  $Nn.7$  bzw.  $Pn.7$  (wobei  $n$  automatisch groß genug berechnet wird).

Ein mit einer mathematischen Funktion (außer VAL) verwendetes Feld kann eine Konstante oder ein Skalar sein und muss numerisches, gepackt numerisches, Ganzzahl- oder Gleitkomma-Format (N, P, I oder F) haben.

Ein mit der Funktion VAL verwendetes Feld kann eine Konstante, ein Skalar oder ein Array sein und muss alphanumerisches Format haben.

**Beispiel für mathematische Funktionen:**

```

** Example 'MATHEX': Mathematical functions
*****
DEFINE DATA LOCAL
1 #A      (N2.1) INIT <10>
1 #B      (N2.1) INIT <-6.3>
1 #C      (N2.1) INIT <0>
1 #LOGA   (N2.6)
1 #SQRTA  (N2.6)
1 #TANA   (N2.6)
1 #ABS    (N2.1)
1 #FRAC   (N2.1)
1 #INT    (N2.1)
1 #SGN    (N1)
END-DEFINE
*
COMPUTE #LOGA = LOG(#A)
WRITE NOTITLE '=' #A 5X 'LOG'          40T #LOGA
*
COMPUTE #SQRTA = SQRT(#A)
    
```

```

WRITE      '=' #A 5X 'SQUARE ROOT' 40T #SQRTA
*
COMPUTE #TANA = TAN(#A)
WRITE      '=' #A 5X 'TANGENT'      40T #TANA
*
COMPUTE #ABS  = ABS(#B)
WRITE      // '=' #B 5X 'ABSOLUTE'  40T #ABS
*
COMPUTE #FRAC = FRAC(#B)
WRITE      '=' #B 5X 'FRACTIONAL'  40T #FRAC
*
COMPUTE #INT  = INT(#B)
WRITE      '=' #B 5X 'INTEGER'      40T #INT
*
COMPUTE #SGN  = SGN(#A)
WRITE      // '=' #A 5X 'SIGN'      40T #SGN
*
COMPUTE #SGN  = SGN(#B)
WRITE      '=' #B 5X 'SIGN'        40T #SGN
*
COMPUTE #SGN  = SGN(#C)
WRITE      '=' #C 5X 'SIGN'        40T #SGN
*
END                                     ↵
    
```

#### Ausgabe des Programms MATHEX:

```

#A:  10.0    LOG                2.302585
#A:  10.0    SQUARE ROOT        3.162277
#A:  10.0    TANGENT             0.648360

#B:  -6.3    ABSOLUTE           6.3
#B:  -6.3    FRACTIONAL         -0.3
#B:  -6.3    INTEGER            -6.0

#A:  10.0    SIGN                1
#B:  -6.3    SIGN                -1
#C:   0.0    SIGN                0      ↵
    
```



# II

## Verschiedene Systemfunktionen

---

Folgende Themen werden behandelt:

**\*MINVAL/\*MAXVAL - Minimum/Maximum auswerten**

**\*TRANSLATE - Zeichen in Groß-/Kleinschreibung umsetzen**

**\*TRIM - Führende und/oder nachfolgende Leerstellen entfernen**

**POS - Feldidentifikationsfunktion**

**RET - Returncode-Funktion**

**SORTKEY - Sortierschlüssel-Funktion**



# 4 \*MINVAL/\*MAXVAL - Minimal-/Maximalwertes eines Feldes

---

- Funktion ..... 26
- Einschränkungen ..... 26
- Syntax-Beschreibung ..... 26
- Tabelle für die Konvertierung der resulting-format/length-Werte ..... 28
- Berechnung der result-format-length ..... 30

```
{ *MINVAL } (([IR=result-format/length]) operand,...)
{ *MAXVAL }
```

*format/length*: Format und Länge können entweder explizit mit der IR-Klausel angegeben oder automatisch mit Hilfe den unten aufgeführten Tabellen errechnet werden, siehe [Tabelle für die Konvertierung der resulting-format/length-Werte](#).

## Funktion

---

Die Natural-Systemfunktion \*MINVAL bzw. \*MAXVAL wertet die Minimal-/Maximalwerte aller gegebenen Operanden aus. Das Ergebnis ist immer ein Skalarwert. Wird ein Array als Operand angegeben, dann wird der Minimal- bzw. Maximalwert aller Array-Felder ausgewertet.

Wenn bei Verwendung von alphanumerischen oder binären Daten dieselben Daten als Argument angegeben sind, (z.B. \*MINVAL('AB', 'AB')), dann ist das Ergebnis das Argument mit dem kleinsten/größten Längenwert.

## Einschränkungen

---

Für die Verwendung der Systemfunktion \*MINVAL/\*MAXVAL gelten folgende Einschränkungen:

- \*MINVAL/\*MAXVAL darf nicht an Stellen verwendet werden, an denen eine Zielvariable erwartet wird.
- \*MINVAL/\*MAXVAL darf nicht in einer anderen Systemfunktion verschachtelt werden

## Syntax-Beschreibung

---

Operanden-Definitionstabelle:

| Operand        | Possible Structure | Possible Formats  | Referencing Permitted | Dynamic Definition |
|----------------|--------------------|-------------------|-----------------------|--------------------|
| <i>operand</i> | C S A G            | A U N P I F B D T | yes                   | no                 |

Syntax-Elementbeschreibung:



| Schlüsselwort               | Beschreibung   |
|-----------------------------|--|
| *MINVAL                     | Berechnet den Minimalwert aller gegebenen Operandenwerte.  |
| *MAXVAL                     | Berechnet den Maximalwert aller gegebenen Operandenwerte.  |
| <i>operand</i>              | Der Operand bzw. die Operanden, deren Minimal-/Maximalwert(e) von der Systemfunktion *MINVAL bzw. *MAXVAL errechnet werden soll. |
| <i>result-format-length</i> | IR-Klausel zur <b>expliziten Angabe</b> der resultierenden Format/Länge-Werte. Siehe unten.                                      |

## IR-Klausel

Diese Klausel kann zur expliziten Angabe der *result-format/length*-Werte für die gesamte Systemfunktion \*MINVAL bzw. \*MAXVAL verwendet werden.

IR=*result-format/length*

$$IR = \left\{ \begin{array}{l} \textit{format-length} \\ ( \left\{ \begin{array}{l} A \\ U \\ B \end{array} \right\} ) \textit{DYNAMIC} \end{array} \right\}$$

Eine Aufstellung der gültigen *result-format/length*-Werte ist im Abschnitt **Tabelle für die Konvertierung der *resulting-format/length*-Werte** enthalten.

Syntax-Elementbeschreibung:

| Schlüsselwort        | Beschreibung   |
|----------------------|--|
| <i>format-length</i> | Der Compiler versucht, die resultierenden Format-/Längenwerte der Gesamtfunktion zu bestimmen. Wenn dies nicht ohne Einbußen an Genauigkeit möglich ist, muss der Programmierer die <i>format-length</i> -Werte mit Hilfe der IR-Operandenerweiterung angeben. |
| A, U oder B          | Format: Alphanumerisch oder binär für dynamische Variable.   |
| DYNAMIC              | Anstelle von festen Format-/Längenwerten kann auch ein alphanumerisches, Unicode- oder binäres Format mit dynamischer Länge angegeben werden.  |

Beispiel:

```
DEFINE DATA LOCAL
1 #RESULTI      (I4)
1 #RESULTA      (A20)
1 #RESULTADYN   (A) DYNAMIC
1 #A(I4)        CONST <1234>
1 #B(A20)       CONST <H'30313233'> /* '0123' stored
1 #C(I2/1:3)    CONST <2000, 2100, 2200>
END-DEFINE
*
#RESULTA      := *MAXVAL((IR=A20)      #A, #B)      /*no error, I4->A20 is allowed!
#RESULTADYN   := *MAXVAL((IR=(A)DYNAMIC) #A, #B)      /*result is (A) dynamic
/* #RESULTI   := *MAXVAL((IR=I4)      #A, #B)      /*compiler error, because conv. ←
A20->I4 is not allowed!
#RESULTI      := *MAXVAL((IR=I4)      #A, #C(*)) /*maximum of the array is ←
evaluated
DISPLAY #RESULTA #RESULTADYN (AL=10) #RESULTI
END
```

## Tabelle für die Konvertierung der resulting-format/length-Werte

---

Es gibt zwei Arten, die resultierenden Format-/Längenwerte für die gesamte Systemfunktion \*MINVAL/\*MAXVAL anzugeben.

- Explizite Angabe der resultierenden Format-/Längenwerte
- Implizite Angabe der resultierenden Format-/Längenwerte

### Explizite Angabe der resultierenden Format-/Längenwerte

Die resultierenden Format-/Längenwerte für die gesamte Systemfunktion \*MINVAL bzw. \*MAXVAL können mit der IR-Klausel angegeben werden. Alle angegebenen Operanden werden in die mit dieser Klausel festgelegten resultierenden Format-/Längenwerte umgewandelt, wenn dies ohne Einbußen bezüglich der Genauigkeit möglich ist. Anschließend erfolgt die Berechnung der Minimal- bzw. Maximalwerte aller umgewandelten Operanden, und es wird ein einziger Skalarwert mit dem errechneten Format-/Längenwert als Ergebnis für die gesamte Systemfunktion festgelegt.

### Implizite Angabe der resultierenden Format-/Längenwerte

Wenn keine IR-Klausel in der Systemfunktion \*MINVAL bzw. \*MAXVAL verwendet wird, erfolgt die Berechnung der resultierenden Format-/Längenwerte anhand der Formate/Längen aller in der Systemfunktion als Argument angegebenen Operanden. Dabei wird der *format/length*-Wert jedes einzelnen Operanden genommen und mit dem *format/length*-Wert des darauf folgenden Operanden in der Argument-Liste zusammengefasst. Der resultierenden Format-/Längenwert von zwei einzelnen Operanden wird dann gemäß der unten aufgeführten Tabellen errechnet.

Die Tabelle für die Konvertierung der *resulting-format/length*-Werte ist in zwei Untertabellen aufgeteilt. Kombinationen, die nicht in diesen beiden Tabellen aufgeführt sind, sind ungültig und

dürfen in der Argument-Liste der Systemfunktion \*MINVAL bzw. \*MAXVAL nicht verwendet werden. Das Schlüsselwort FLF gibt an, wann zur Vermeidung von Ungenauigkeiten die IR-Klausel zur Festlegung des resultierenden Format-/Längenwertes benutzt werden muss.

**Tabelle 1**

Enthält alle numerischen Kombinationsmöglichkeiten für zwei verschiedene Operanden.

|                | Zweiter Operand |             |             |              |   |   |
|----------------|-----------------|-------------|-------------|--------------|---|---|
|                | Format/Länge    | I1          | I2          | I4           | Pa . b, Na . b  | F4, F8                                  |
| Erster Operand | I1              | I1          | I2          | I4           | Pmax(3,a).b   | F8                                      |
|                | I2              | I2          | I2          | I4           | Pmax(5,a).b   | F8                                      |
|                | I4              | I4          | I4          | I4           | Pmax(10,a).b  | F8                                      |
|                | Px . y, Nx . y  | Pmax(3,x).y | Pmax(5,x).y | Pmax(10,x).y | wenn max(x, a) + max(y, b) <= 29<br>Pmax(x, a).max(y, b)<br>sonst FLF | wenn y=0 und x <=15;<br>F8<br>sonst FLF |
|                | F4, F8          | F8          | F8          | F8           | wenn b=0 und a <=15<br>F8<br>sonst FLF                                | F8                                      |

Legende:

|  |   |
|--|---|
| <b>FLF</b>                             | Format-/Längenangabe mittels IR-Klausel zwingend erforderlich.  |
| <b>I<sub>x</sub></b>                   | Format/Länge ist Integer (Ganzzahl). <i>x</i> gibt die Anzahl der Bytes an, die zum Speichern des Integer-Wertes benutzt werden.  |
| <b>F<sub>x</sub></b>                   | Format/Länge ist Float (Gleitkomma). <i>x</i> gibt die Anzahl der Bytes an, die zum Speichern des Float-Wertes benutzt werden.  |
| <b>P<sub>x.y</sub></b><br><b>Pa, b</b> | Format ist Packed (gepackt) mit entsprechender Anzahl an Stellen vor dem Dezimalpunkt ( <i>x, a</i> ) und der Genauigkeit ( <i>y, b</i> ).  |
| <b>N<sub>x.y</sub></b><br><b>Na, b</b> | Format ist Numeric (numerisch) mit entsprechender Anzahl an Stellen vor dem Dezimalpunkt ( <i>x, a</i> ) und der Genauigkeit ( <i>y, b</i> ).   |
| <b>Pmax(c, d).e</b>                    | Das resultierende Format ist Packed (gepackt). Die Berechnung der Länge erfolgt anhand der nachfolgenden Informationen. Die Anzahl an Stellen vor dem Dezimalpunkt ist der Maximalwert von <i>c</i> und <i>d</i> . Der Genauigkeitswert ist <i>e</i> .                                  |
| <b>Pmax(c, d).max(e, f)</b>            | Das resultierende Format ist Packed (gepackt). Die Berechnung der Länge erfolgt anhand der nachfolgenden Informationen. Die Anzahl an Stellen vor dem Dezimalpunkt ist der Maximalwert von <i>c</i> und <i>d</i> . Der Genauigkeitswert ist der Maximalwert von <i>e</i> und <i>f</i> . |

**Tabelle 2**

Enthält alle Formate und Längen, die für Operanden in den Systemfunktionen \*MINVAL/\*MAXVAL verwendet werden können.

|                | Zweiter Operand |    |    |               |               |               |
|----------------|-----------------|----|----|---------------|---------------|---------------|
|                | Format-length   | D  | T  | Aa, A dynamic | Ba, B dynamic | Ua, U dynamic |
| Erster Operand | D               | D  | T  | NA            | NA            | NA            |
|                | T               | T  | T  | NA            | NA            | NA            |
|                | Ax, A dynamic   | NA | NA | A dynamic     | A dynamic     | U dynamic     |
|                | Bx, B dynamic   | NA | NA | A dynamic     | B dynamic     | U dynamic     |
|                | Ux, U dynamic   | NA | NA | U dynamic     | U dynamic     | U dynamic     |

Legende:

|                  |  |
|------------------|--|
| <b>NA</b>        | Unzulässige Kombination.                       |
| <b>D</b>         | Datenformat.                                   |
| <b>T</b>         | Zeitformat.                                    |
| <b>Bx, Ba</b>    | Binärformat mit Länge x, a.                    |
| <b>Ax, Aa</b>    | Alphanumerisches Format mit Länge x, a.        |
| <b>Ux, Ua</b>    | Unicode-Format mit Länge x, a.                 |
| <b>B dynamic</b> | Binärformat mit dynamischer Länge.             |
| <b>A dynamic</b> | Alphanumerisches Format mit dynamischer Länge. |
| <b>U dynamic</b> | Unicode-Format mit dynamischer Länge.          |

## Berechnung der result-format-length

Anhand der oben aufgeführten Regeln ist der Compiler in der Lage, die Source-Operanden unter Berücksichtigung von Operandenpaaren zu verarbeiten und die Zwischenergebnisse für jedes Paar zu berechnen. Das erste Paar besteht aus dem ersten und dem zweiten Operand, das zweite Paar aus dem Zwischenergebnis und dem dritten Operand usw. Nach Verarbeitung aller Operanden stellt das letzte Ergebnis den Vergleich zwischen Format und Länge dar, die zum Vergleich mit allen Operanden zwecks Berechnung des Minimal- bzw. Maximalwertes verwendet werden. Bei Verwendung dieser Format-/Längenberechnungsmethode können die Operanden in beliebiger Reihenfolge auftreten.

Beispiel:

```

DEFINE DATA LOCAL
1 A (I2)      INIT <34>
1 B (P4.2)    INIT <1234.56>
1 C (N4.4)    INIT <12.6789>
1 D (I1)      INIT <100>
1 E (I4/1:3)  INIT <32, 6745, 456>
1 #RES-MIN (P10.7)
1 #RES-MAX (P10.7)
END-DEFINE
*
MOVE *MINVAL(A, B, C, D, E(*)) TO #RES-MIN
MOVE *MAXVAL(A, B, C, D, E(*)) TO #RES-MAX
DISPLAY #RES-MIN #RES-MAX
END
    
```

Ausgabe:

```

#RES-MIN                                     #RES-MAX
-----
12.6789000                                6745.0000000
    
```

Die folgende Tabelle zeigt die einzelnen Schritte zur automatischen Format-/Längenberechnung im obigen Beispiel. Sie enthält die Zwischenergebnisse (ir) aller Schritte und den Format-/Längenwert des Vergleichs (cf), der als *result-format/length*-Wert verwendet wird.

| Berechnungsreihenfolge | Name des ersten Operanden | Format/Länge des ersten Operanden bzw. Zwischenergebnis (ir) | Name des zweiten Operanden | Format/Länge des zweiten Operanden bzw. Zwischenergebnis (ir) | Format/Länge des Zwischenergebnisses(ir) |
|------------------------|---------------------------|--|----------------------------|---|--|
| 1.                     | A                         | I2   | B                          | P4.2  | ir1 = P5.2                               |
| 2.                     | ir1                       | P5.2   | C                          | N4.4  | ir2 = P5.4                               |
| 3.                     | ir2                       | P5.4   | D                          | I1  | ir3 = P5.4                               |
| 4.                     | ir3                       | P5.4   | E                          | I4  | <b>cf = P10.4</b>                        |

Zur Laufzeit werden alle Operanden in die cf-Format/Länge umgewandelt. Danach werden alle umgewandelten Werte miteinander verglichen und der entsprechende Minimal- bzw. Maximalwert wird errechnet.

**Anmerkungen:**

1. Wenn nur ein einzelner Operand angegeben wird, hat *result-format-length* der Format-/Längenwert dieses Operanden.
2. Wenn ein binärer Operand mit einer Länge im Bereich von 1 - 4 als Argument innerhalb der Systemfunktion \*MINVAL/\*MAXVAL zusammen mit einem alphanumerischen oder einem Unicode-

Operanden angegeben wird, dann wird das Zwischenergebnis (*result-format-length*) als alphanumerisches bzw. als Unicode-Format mit dynamischer Länge bewertet.

In diesem Fall wird der Wert des binären Operanden als numerischer Wert betrachtet, der gemäß den Datenübertragungsregeln in das *result-format-length* umgewandelt wird (der binäre numerische Wert wird in das Format Upacked gewandelt), bevor der Minimal-/Maximalwert berechnet wird.

Beispiel:


```
DEFINE DATA LOCAL
1 #B4 (B4) INIT <1>
1 #A10(A10) INIT <"2">
END-DEFINE
WRITE "=" *MAXVAL(#A10, #B4) (AL=60) /* RESULT FORMAT-LENGTH IS (A)DYNAMIC: "2"
WRITE "=" *MINVAL(#A10, #B4) (AL=60) /* RESULT FORMAT-LENGTH IS (A)DYNAMIC: "1"
END
```

Der Zwischenwert *result-format-length* (#A10, #B4) ist A dynamisch.

Daher wird zunächst #A10 nach A dynamisch gewandelt, ebenso wie #B4 nach A dynamisch gewandelt wird (unter Berücksichtigung der Datenübertragungsregeln), bevor das Zwischenergebnis der beiden Operanden berechnet wird.

### Reihenfolge der Berechnung von Format und Länge

Die folgende Grafik zeigt die Reihenfolge, in der die Berechnung von Format und Länge erfolgt:



Legende:

|                      |  |
|----------------------|--|
| <b>ir1, ir2, ir3</b> | Zwischenergebnis (intermediate result) 1, 2, 3.              |
| <b>cf</b>            | Resultierende Vergleich-Format- und Längewerte (comparison). |





# 5

## \*TRANSLATE - Umsetzung in Groß-/Kleinschreibung

---

|                             |    |
|-----------------------------|----|
| ▪ Funktion .....            | 36 |
| ▪ Einschränkungen .....     | 36 |
| ▪ Syntax-Beschreibung ..... | 36 |
| ▪ Beispiel .....            | 37 |

```
*TRANSLATE ( operand [ , { LOWER } ] )
```

Format/Länge: wie bei *operand*.

## Funktion

Die Systemfunktion \*TRANSLATE dient zum Umsetzen von Zeichen eines Operanden mit Alphanumerischem oder binärem Format in Groß- oder Kleinbuchstaben. Der Inhalt des Operanden bleibt dabei unverändert.

\*TRANSLATE kann als *operand* an jeder Stelle eines Statements angegeben werden, an der ein Operand mit Format A oder B zulässig ist.

## Einschränkungen

Für die Verwendung der Systemfunktion \*TRANSLATE gelten folgende Einschränkungen:

- \*TRANSLATE darf nicht an Stellen verwendet werden, an denen eine Zielvariable erwartet wird.
- \*TRANSLATE darf nicht in einer anderen Systemfunktion verschachtelt werden.

## Syntax-Beschreibung

Operanden-Definitionstabelle:

| Operand        | Mögliche Struktur | Mögliche Formate | Referenzierung erlaubt | Dynam. Definition |
|----------------|-------------------|------------------|------------------------|-------------------|
| <i>operand</i> | C S A             | A U B            | ja                     | nein              |

Syntax-Elementbeschreibung:

|                                      |   |
|--------------------------------------|---|
| *TRANSLATE ( <i>operand</i> , LOWER) | <b>Umsetzung in Kleinbuchstaben</b><br>Bei Angabe des Schlüsselworts LOWER als zweites Argument angegeben wird die Zeichenkette in <i>operand</i> in Kleinbuchstaben umgesetzt. |
| *TRANSLATE ( <i>operand</i> , UPPER) | <b>Umsetzung in Großbuchstaben</b><br>Bei Angabe des Schlüsselworts UPPER als zweites Argument angegeben wird die Zeichenkette in <i>operand</i> in Großbuchstaben umgesetzt.   |

## Beispiel

```
DEFINE DATA LOCAL
1 #SRC (A)DYNAMIC INIT <'aBcDeFg !$%&/()=?'>
1 #DEST (A)DYNAMIC
END-DEFINE
*
PRINT 'Source string to be translated:.....' #SRC
*
MOVE *TRANSLATE(#SRC, UPPER) TO #DEST
PRINT 'Source string translated into upper case:' #DEST
*
MOVE *TRANSLATE(#SRC, LOWER) TO #DEST
PRINT 'Source string translated into lower case:' #DEST
END
```

### Ausgabe:

```
Source string to be translated:..... aBcDeFg !$%&/()=?
Source string translated into upper case: ABCDEFG !$%&/()=?
Source string translated into lower case: abcdefg !$%&/()=?
```



# 6 \*TRIM - Entfernen von führenden und/oder nachfolgenden

## Leerstellen

---

|                             |    |
|-----------------------------|----|
| ▪ Funktion .....            | 40 |
| ▪ Einschränkungen .....     | 40 |
| ▪ Syntax-Beschreibung ..... | 40 |
| ▪ Beispiel .....            | 41 |

```
*TRIM ( operand [ , { LEADING } ] )
```

Format/Länge: wie bei *operand* (A oder B)/DYNAMIC.

## Funktion

---

Die Systemfunktion \*TRIM dient zum Entfernen von führenden und/oder nachfolgenden Leerstellen aus einer alphanumerischen oder binären Zeichenkette. Der Inhalt von *operand* bleibt dabei unverändert. Bei Verwendung einer dynamischen Variablen als *operand*, wird die Länge dieser Variablen dem Ergebnis entsprechend angepasst.

Die Systemfunktion \*TRIM kann als *operand* an jeder Stelle eines Statements angegeben werden, an der ein Operand mit Format A oder B zulässig ist.

## Einschränkungen

---

Für die Verwendung der Systemfunktion \*TRIM gelten folgende Einschränkungen:

- \*TRIM darf nicht an Stellen verwendet werden, an denen eine Zielvariable erwartet wird.
- \*TRIM darf nicht in einer anderen Systemfunktion verschachtelt werden.
- Wenn der Operand eine statische Variable ist, kann man mit \*TRIM keine führenden Leerstellen entfernen, weil bei statischen Variablen die verbleibenden nachfolgenden Stellen des Variablenspeichers mit Leerzeichen aufgefüllt werden.

## Syntax-Beschreibung

---

Operanden-Definitionstabelle:

| Operand        | Mögliche Struktur |   |   |  | Mögliche Formate |   |   |  |  |  |  |  |  |  |  |  | Referenzierung erlaubt | Dynam. Definition |
|----------------|-------------------|---|---|--|------------------|---|---|--|--|--|--|--|--|--|--|--|------------------------|-------------------|
| <i>operand</i> | C                 | S | A |  | A                | U | B |  |  |  |  |  |  |  |  |  | ja                     | nein              |

Syntax-Elementbeschreibung:

|                                      |  |
|--------------------------------------|--|
| *TRIM( <i>operand</i> ,<br>LEADING)  | <b>Führende Leerstellen entfernen</b><br><br>Bei Angabe des Schlüsselworts LEADING als zweites Argument werden alle führenden Leerstellen aus dem in <i>operand</i> enthaltenen String entfernt.   |
| *TRIM( <i>operand</i> ,<br>TRAILING) | <b>Nachfolgende Leerstellen entfernen</b><br><br>Bei Angabe des Schlüsselworts TRAILING als zweites Argument werden alle nachfolgenden Leerstellen aus dem in <i>operand</i> enthaltenen String entfernt.  |
| *TRIM( <i>operand</i> )              | <b>Führende und Nachfolgende Leerstellen entfernen</b><br><br>Wenn kein zweites Schlüsselwort als Argument angegeben wird, bewirkt TRIM, dass alle führenden und nachfolgenden Leerstellen aus dem in <i>operand</i> enthaltenen String entfernt werden. |

## Beispiel

### Verwendung eines alphanumerischen Arguments

```

DEFINE DATA LOCAL
/*****
/* STATIC VARIABLE DEFINITIONS
/*****
1 #SRC (A15) INIT <' ab CD '>
1 #DEST (A15)

/* FOR PRINT OUT WITH DELIMITERS
1 #SRC-PRN (A20)
1 #DEST-PRN (A20)

/*****
/* DYNAMIC VARIABLE DEFINITIONS
/*****
1 #DYN-SRC (A)DYNAMIC INIT <' ab CD '>
1 #DYN-DEST (A)DYNAMIC

/* FOR PRINT OUT WITH DELIMITERS
1 #DYN-SRC-PRN (A)DYNAMIC
1 #DYN-DEST-PRN (A)DYNAMIC

END-DEFINE

PRINT 'static variable definition:'
PRINT '-----'
COMPRESS FULL ':' #SRC ':' TO #SRC-PRN LEAVING NO SPACE
PRINT ' '
PRINT ' 123456789012345      123456789012345'

MOVE *TRIM(#SRC, LEADING) TO #DEST

```

## \*TRIM - Entfernen von führenden und/oder nachfolgenden Leerstellen

```
COMPRESS FULL ':' #DEST ':' TO #DEST-PRN LEAVING NO SPACE
DISPLAY #SRC-PRN #DEST-PRN '*TRIM(#SRC, LEADING)'
```

```
MOVE *TRIM(#SRC, TRAILING) TO #DEST
COMPRESS FULL ':' #DEST ':' TO #DEST-PRN LEAVING NO SPACE
DISPLAY #SRC-PRN #DEST-PRN '*TRIM(#SRC, TRAILING)'
```

```
MOVE *TRIM(#SRC) TO #DEST
COMPRESS FULL ':' #DEST ':' TO #DEST-PRN LEAVING NO SPACE
DISPLAY #SRC-PRN #DEST-PRN '*TRIM(#SRC)'
```

```
PRINT ' '
PRINT 'dynamic variable definition:'
PRINT '-----'
COMPRESS FULL ':' #DYN-SRC ':' TO #DYN-SRC-PRN LEAVING NO SPACE
PRINT ' '
PRINT ' 1234567890          12345678'
```

```
MOVE *TRIM(#DYN-SRC, LEADING) TO #DYN-DEST
COMPRESS FULL ':' #DYN-DEST ':' TO #DYN-DEST-PRN LEAVING NO SPACE
DISPLAY (AL=20) #DYN-SRC-PRN #DYN-DEST-PRN '*TRIM(#SRC, LEADING)'
```

```
MOVE *TRIM(#DYN-SRC, TRAILING) TO #DYN-DEST
COMPRESS FULL ':' #DYN-DEST ':' TO #DYN-DEST-PRN LEAVING NO SPACE
DISPLAY (AL=20) #DYN-SRC-PRN #DYN-DEST-PRN '*TRIM(#SRC, TRAILING)'
```

```
MOVE *TRIM(#DYN-SRC) TO #DYN-DEST
COMPRESS FULL ':' #DYN-DEST ':' TO #DYN-DEST-PRN LEAVING NO SPACE
DISPLAY (AL=20) #DYN-SRC-PRN #DYN-DEST-PRN '*TRIM(#SRC)'
```

```
PRINT ' '
PRINT '":" := delimiter character to show the start and ending of a string!'
END
```

### Ausgabe:

```
      #SRC-PRN          #DEST-PRN
-----
static variable definition:
-----
123456789012345      123456789012345
: ab CD      :      : ab CD      :      *TRIM(#SRC, LEADING)
: ab CD      :      : ab CD      :      *TRIM(#SRC, TRAILING)
: ab CD      :      : ab CD      :      *TRIM(#SRC)

dynamic variable definition:
-----
1234567890          12345678
: ab CD :      : ab CD :      *TRIM(#SRC, LEADING)
```



```
: ab CD :           : ab CD:           *TRIM(#SRC, TRAILING)
: ab CD :           :ab CD:           *TRIM(#SRC)
':' := delimiter character to show the start and ending of a string!
```



# 7

## POS - Feldidentifikationsfunktion

---

|                  |
|------------------|
| Format/Länge: I4 |
|------------------|

Die Systemfunktion `POS(field-name)` enthält die interne Identifikation des Feldes, dessen Name mit der Systemfunktion angegeben wird.

`POS(field-name)` identifiziert ein bestimmtes Feld, unabhängig von seiner Position in einer Maske (Map). Auch wenn sich die Reihenfolge und Anzahl der Felder in einer Maske ändert, identifiziert `POS(field-name)` nach wie vor eindeutig dasselbe Feld. Damit genügt zum Beispiel ein einziges REINPUT-Statement, um es von der Programmlogik abhängig zu machen, welches Feld MARKiert werden soll.

Beispiel:

```
DECIDE ON FIRST VALUE OF ...
  VALUE ...
    COMPUTE #FIELDX = POS(FIELD1)
  VALUE ...
    COMPUTE #FIELDX = POS(FIELD2)
  ...
END-DECIDE
...
REINPUT ... MARK #FIELDX
```

Wenn das mit `POS` angegebene Feld ein Array ist, muss eine bestimmte Ausprägung angegeben werden; zum Beispiel `POS(FIELDX(5))`. Auf einen Array-Bereich kann `POS` nicht angewendet werden.

### POS und \*CURS-FIELD

`POS(field-name)` kann in Verbindung mit der Natural-Systemvariablen `*CURS-FIELD` dazu verwendet werden, die Ausführung bestimmter Funktionen davon abhängig zu machen, in welchem Feld der Cursor zur Zeit steht.

\*CURS-FIELD enthält die interne Identifikation des Feldes, in dem sich der Cursor zur Zeit befindet; \*CURS-FIELD kann nicht alleine, sondern nur zusammen mit `POS(field-name)` verwendet werden. Sie können beide zusammen dazu benutzen, um zu prüfen, ob sich der Cursor gerade in einem bestimmten Feld befindet, und die weitere Verarbeitung von dieser Bedingung abhängig machen.

Beispiel:

```
IF *CURS-FIELD = POS(FIELDX)
  MOVE *CURS-FIELD TO #FIELDY
END-IF
...
REINPUT ... MARK #FIELDY
```



### Anmerkungen:

1. Die Werte von \*CURS-FIELD und `POS(field-name)` dienen nur zur internen Identifikation der Felder und können nicht für arithmetische Operationen verwendet werden.
2. Der von `POS(field-name)` zurückgegebene Wert für eine Ausprägung eines X-Arrays (ein Array, für das wenigstens eine Dimension als erweiterbar angegeben ist) kann sich ändern, nachdem die Anzahl der Ausprägungen für eine Dimension des Arrays mittels der Statements `EXPAND`, `RESIZE` oder `REDUCE` geändert wurde.
3. Natural RPC: Wenn \*CURS-FIELD und `POS(field-name)` sich auf eine Kontextvariable beziehen, können die daraus resultierenden Informationen nur innerhalb derselben Konversation verwendet werden.
4. In Natural for Ajax-Anwendungen dient \*CURS-FIELD zur Identifikation des Operanden, welcher den Wert des Control darstellt, welches den Eingabefokus hat. Sie können \*CURS-FIELD in Verbindung mit der POS-Funktion benutzen, um eine Prüfung auf das Control, welches den Eingabefokus hat, zu veranlassen und die Verarbeitung in Abhängigkeit von diesem Zustand durchzuführen.

Siehe auch

- *Dialog-Gestaltung, Feld-sensitive Verarbeitung und Einfachere Programmierung im Natural Leitfaden zur Programmierung.*

# 8 RET - Returncode-Funktion

---

Format/Länge: I4

Die Systemfunktion `RET(program-name)` kann dazu verwendet werden, den Returncode eines nicht in Natural geschriebenen Programms, das über ein `CALL`-Statement aufgerufen wurde, zu erhalten.

`RET(program-name)` kann in einem `IF`-Statement sowie in den Arithmetik-Statements `ADD`, `COMPUTE`, `DIVIDE`, `MULTIPLY` und `SUBTRACT` verwendet werden.

Beispiel:

```
DEFINE DATA LOCAL
1 #RETURN (I4)
...
END-DEFINE
...
...
CALL 'PROG1'
IF RET('PROG1') > #RETURN
    WRITE 'ERROR OCCURRED IN PROGRAM 1'
END-IF
...
```



## 9 SORTKEY - Sort-Key Function

---

SORTKEY (*character-string*)

Diese Systemfunktion dient zum Konvertieren von „nicht richtig sortierten“ Zeichen (oder Kombinationen von Zeichen) in andere Zeichen (oder Kombinationen von Zeichen), die vom Sortierprogramm oder Datenbanksystem alphabetisch „richtig sortiert“ werden.

Format/Länge: A253

Es gibt in vielen Landessprachen Zeichen (oder Zeichenkombinationen), die von einem Sortierprogramm oder Datenbanksystem nicht in der richtigen alphabetischen Reihenfolge sortiert werden, da die Reihenfolge der Zeichen im vom Computer verwendeten Zeichensatz nicht immer der alphabetischen Reihenfolge der Zeichen entspricht.

Zum Beispiel wird der spanische Buchstabe CH in der Regel von einem Sortierprogramm bzw. Datenbanksystem wie zwei Buchstaben behandelt und zwischen CG und CI einsortiert — gehört aber eigentlich als eigener Buchstabe im spanischen Alphabet zwischen "C" und "D".

Oder es kann sein, dass Kleinbuchstaben und Großbuchstaben entgegen Ihren Wünschen bei der Sortierreihenfolge nicht gleich behandelt werden, dass Ziffern vor Buchstaben sortiert werden (Sie aber wünschen, dass Buchstaben vor Ziffern sortiert werden) oder dass Sonderzeichen (z.B. Bindestriche in Doppelnamen) zu einer unerwünschten Sortierreihenfolge führen.

In solchen Fällen können Sie die Systemfunktion SORTKEY(*character-string*) benutzen. Die von SORTKEY berechneten Werte werden nur als Sortierkriterium benutzt, während die ursprünglichen Werte für die Interaktion mit dem Endbenutzer verwendet werden.

Sie können die SORTKEY-Funktion in einem COMPUTE sowie in einer logischen Bedingung als arithmetischen Operanden verwenden.

Als *character-string* können Sie eine alphanumerische Konstante oder Variable oder eine einzelne Ausprägung eines alphanumerischen Arrays angeben.

Wenn Sie die SORTKEY-Funktion in einem Natural-Programm angeben, wird der User-Exit NATUSK $nn$  aufgerufen — wobei  $nn$  der aktuelle Sprachcode (d.h. der aktuelle Wert der Systemvariablen \*LANGUAGE) ist.

Sie können diesen User-Exit in jeder Programmiersprache, die über eine Standard-CALL-Schnittstelle verfügt, schreiben. Der mit SORTKEY angegebene *character-string* wird an den User-Exit übergeben. Der User-Exit muss so programmiert werden, dass er „falsch sortierte“ Zeichen in dieser Zeichenkette in entsprechende „richtig sortierte“ Zeichen umsetzt. Die umgesetzte Zeichenkette wird dann vom Natural-Programm zur weiteren Verarbeitung verwendet.

Allgemeine Aufruf-Konventionen für externe Programme sind in der Dokumentation zum CALL-Statement erläutert.

Nähere Informationen zu den Aufruf-Konventionen für SORTKEY User-Exits finden Sie im Abschnitt *User Exit for Computation of Sort Keys* in der *Natural Operations Documentation*.

Beispiel:

```
DEFINE DATA LOCAL
1 CUST VIEW OF CUSTOMERFILE
  2 NAME
  2 SORTNAME
END-DEFINE
...
*LANGUAGE := 4
...
REPEAT
  INPUT NAME
  SORTNAME := SORTKEY(NAME)
  STORE CUST
  END TRANSACTION
  ...
END-REPEAT
...
READ CUST BY SORTNAME
  DISPLAY NAME
END-READ
...
```

Angenommen, im obigen Beispiel würden bei mehrmaliger Ausführung des INPUT-Statements nacheinander die Werte "Sanchez", "Sandino" und "Sancinto" eingegeben.

Bei der Zuweisung von SORTKEY(NAME) zu SORTNAME würde der User-Exit NATUSK04 aufgerufen. Dieser müsste so programmiert werden, dass er zunächst alle Kleinbuchstaben in Großbuchstaben umsetzt und dann die Zeichenfolge "CH" in "C $x$ " umsetzt — wobei  $x$  dem letzten Zeichen im verwendeten Zeichensatz entspricht, also hexadezimal H'FF' (vorausgesetzt dieses letzte Zeichen ist kein druckbares Zeichen).



Es werden sowohl die „eigentlichen“ Namen (NAME) als auch die für die gewünschte Sortierung umgesetzten Namen (SORTNAME) gespeichert. Zum Lesen der Datei wird SORTNAME verwendet. Dann würden bei Ausführung des DISPLAY-Statements die Namen in der richtigen spanischen alphabetischen Reihenfolge ausgegeben:

```
Sancinto  
Sanchez  
Sandino
```



# III Funktionen, die als Natural-Objekte ausgeliefert werden

---



# 10 Funktionen, die als Natural-Objekte ausgeliefert werden

---

- URL-Kodierung ..... 56
- Base64-Kodierung ..... 67

Dieses Dokument beschreibt Funktionen, die unter Verwendung von Natural-Objekten des Typs Function implementiert wurden.

Diese Function-Objekte (und ihre Prototyp-Definitionen), deren Namen mit SAG beginnen, werden in der Natural-Library SYSTEM in der Systemdatei FNAT ausgeliefert. Beispiele für Function Calls sind in der Library SYSEXPG vorhanden.

Weitere Informationen siehe *Function Call* im *Leitfaden zur Programmierung*.

## URL-Kodierung

---

Bei HTTP-Anfragen über Natural-Anwendungsschnittstellen ist es oftmals notwendig, dass der URI (Uniform Resource Identifier) URL-kodiert ist. Das Statement REQUEST DOCUMENT benötigt eine solche URL, um auf ein Dokument zugreifen zu können.

URL-Kodierung, auch Prozentkodierung genannt, ist ein Mechanismus, um bestimmte Sonderzeichen in Teilen einer URL zu ersetzen. Zur Bildung einer URL werden nur Zeichen des US-ASCII-Zeichensatzes verwendet. Einige Zeichen des US-ASCII-Zeichensatzes haben bei Verwendung in einer URL eine besondere Bedeutung - sie sind als „reservierte“ Steuerzeichen eingestuft, die zur Strukturierung der Zeichenkette in verschiedene semantische Unterkomponenten dienen. Der Quasi-Standard bezüglich der generischen Syntax einer URL ist im Dokument RFC3986 enthalten, das von der Internet-Community zusammengestellt worden ist. Darin wird beschrieben, unter welchen Bedingungen eine URL-Kodierung notwendig ist. Dazu gehört auch die Darstellung von Zeichen, welche nicht Bestandteil des US-ASCII-Zeichensatzes sind (zum Beispiel das Euro-Zeichen) und die Verwendung der reservierten Zeichen.

Reservierte Zeichen sind:

|   |   |   |   |   |    |   |   |   |   |   |   |   |   |   |   |   |   |   |
|---|---|---|---|---|----|---|---|---|---|---|---|---|---|---|---|---|---|---|
| ? | = | & | # | ! | \$ | % | ' | ( | ) | * | + | , | / | : | ; | @ | [ | ] |
|---|---|---|---|---|----|---|---|---|---|---|---|---|---|---|---|---|---|---|

Nicht reservierte Zeichen sind:

|     |     |     |   |   |   |   |
|-----|-----|-----|---|---|---|---|
| A-Z | a-z | 0-9 | - | _ | . | ~ |
|-----|-----|-----|---|---|---|---|

Eine URL besteht aus reservierten und nicht-reservierten Zeichen; andere Zeichen dürfen in ihr nicht vorkommen. Wenn andere Byte-Werte benötigt werden (die keinem der reservierten und nicht reservierten Zeichen entsprechen) oder wenn reservierte Zeichen als Daten verwendet werden (die keine besondere semantische Bedeutung im URL-Kontext haben sollen), müssen diese Werte in die „Prozentkodierung“ konvertiert werden: ein Prozentzeichen und direkt danach die aus zwei Zeichen bestehende hexadezimale Darstellung des Codepoint (bedingt durch das Kodierschema Windows-1252). Das hat zur Folge, dass ein Pluszeichen (+) als %2B, ein Prozentzeichen als %25 und ein „at“-Zeichen (@) als %40 in der Zeichenkette erscheint.

Die nachfolgend beschriebenen Kodierfunktionen bearbeiten die kompletten Eingabe-Zeichenkette. Bitte beachten Sie, dass Sie keine komplette URL oder Teile davon kodieren, falls diese Steuerzeichen (reservierte Zeichen) enthalten, die nicht in die Prozentform umgesetzt werden dürfen. Die hier beschriebenen Funktionen sollten nur bei Zeichen angewendet werden, welche nicht für den Gebrauch in einer URL erlaubt sind, und bei Zeichen mit einer besonderen Bedeutung im URL-Kontext, die als Datenelement übergeben werden.

- Einfache Kodierung
- SAGENC - Einfache Kodierung (Format A nach Format A)
- SAGDEC - Einfache Dekodierung (Format A nach Format A)
- Erweiterte Kodierung
- SAGENCE - Erweiterte Kodierung (Format U nach Format A, optionale Parameter)
- SAGDECE - Erweiterte Dekodierung (Format A nach Format U, optionale Parameter)
- Beispiel-Programm

## Einfache Kodierung

Der einzelne Eingabeparameter enthält die zu kodierende oder zu dekodierende Zeichenkette. Alle darin enthaltenen Daten werden so betrachtet, als ob sie in der Codepage EBCDIC 1140 dargestellt sind, und zwar unabhängig davon welche Session-Codepage zurzeit aktiv ist. Zur Ausführung der Funktion SAGENC bzw. SAGDEC ist keine Unicode-Unterstützung erforderlich. Die folgenden Zeichen werden durch die entsprechenden hexadezimalen US-ASCII-Zeichen ersetzt.

| Das Zeichen,...                        | <  | (  | +  |    | &  | !  | \$ | *  | )  | ;  | /  | ,  | %  | >  | ?  | `  | :  | #  | @  | '  | =  | "  | ^  | [  | ]  | {  | }  | \  |
|--|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|----|
| das sich am EBCDIC Codepoint befindet, | 4C | 4D | 4E | 4F | 50 | 5A | 5B | 5C | 5D | 5E | 61 | 6B | 6C | 6E | 6F | 79 | 7A | 7B | 7C | 7D | 7E | 7F | B0 | BA | BB | C0 | D0 | E0 |
| wird kodiert zu %nn                    | 3C | 28 | 2B | 7C | 26 | 21 | 24 | 2A | 29 | 3B | 2F | 2C | 25 | 3E | 3F | 60 | 3A | 23 | 40 | 27 | 3D | 22 | 5E | 5B | 5D | 7B | 7D | 5C |

Folgende Funktionen stehen zur Verfügung:

- SAGENC - Einfache Kodierung (Format A nach Format A)
- SAGDEC - Einfache Dekodierung (Format A nach Format A)

### SAGENC - Einfache Kodierung (Format A nach Format A)

Die Funktion SAGENC kodiert eine Zeichenkette in die prozentkodierte Form. Gemäß Standard RFC3986 werden reservierte Zeichen und Zeichen unterhalb von US-ASCII x'7F' (die in einer URL nicht erlaubt sind) prozentkodiert; ein Leerzeichen wird durch ein Pluszeichen (+) ersetzt. Nicht reservierte Zeichen gemäß RCF3986 und Zeichen oberhalb von US-ASCII x'7F', zum Beispiel deutsche Umlaute, werden nicht kodiert. Falls Sie solche Zeichen kodieren möchten, müssen Sie die erweiterte Kodierfunktion SAGENCE.

| Objekt   | Beschreibung  |
|----------|---|
| SAGENC   | Dies ist der Aufruf der einfachen Kodierfunktion.   |
| SAGENCPC | Der Copycode, der die Prototyp-Definition enthält, wird nur bei der Kompilierung benutzt, um den Typ der Rückgabewaren als Referenz für den Funktionsaufruf zu bestimmen und um die Parameter zu prüfen, falls dies gewünscht wird.<br><br>SAGENCPC ist optional. |
| URLX01   | <b>Beispiel-Programm</b> in der Library SYSEXP.<br><br><pre>#URL-ENC := SAGENC(&lt;#URL-DEC&gt;)</pre>  |

### SAGDEC - Einfache Dekodierung (Format A nach Format A)

Die Funktion SAGDEC dient zum Dekodieren der von der Funktion SAGENC gelieferten Prozentkodierungen. Außer der zu dekodierenden Zeichenkette werden keine weiteren Parameter benötigt.

| Objekt  | Beschreibung   |
|---------|--|
| SAGDEC  | Dies ist der Aufruf der einfachen Dekodierfunktion.  |
| SAGDECP | Der Copycode, der die Prototyp-Definition enthält, wird nur bei der Kompilierung benutzt, um den Typ der Rückgabewaren als Referenz für den Funktionsaufruf zu bestimmen und um die Parameter zu prüfen, falls dies gewünscht wird.<br><br>SAGDECP ist optional. |
| URLX01  | <b>Beispiel-Programm</b> in der Library SYSEXP.<br><br><pre>#URL-DEC := SAGDEC(&lt;#URL-ENC&gt;)</pre>   |



## Erweiterte Kodierung

Die erweiterte Funktion berücksichtigt alle in der Spezifikation RFC3986 aufgeführten Sonderfälle. Die folgenden Parameter können berücksichtigt werden (Die Standardvorgabewerte sind fett hervorgehoben.):

1. `<dynamic U-string>` soll kodiert/dekodiert werden.
2. Return Code:  $\diamond 0$  (Natural-Fehler) bei Auftreten eines Fehlers im `MOVE ENCODED`-Statement.
3. Fehlerzeichen falls Return Code  $\diamond 0$ .
4. Leerzeichen: `%20`/+/nicht kodieren (Standardeinstellung: **+**)
5. Nicht reservierte Zeichen: kodieren/**nicht kodieren**
6. Reservierte Zeichen: **kodieren**/nicht kodieren
7. Sonstige Sonderzeichen (weder nicht reservierte noch reservierte Zeichen): **kodieren**/nicht kodieren
8. Zeichen-Prozentkodierung: ISO-8859-1/UTF-8/beliebige andere Codepage/falls = ' ' dann \*CODEPAGE (Natural-Standard-Codepage, nicht die Codepage für die Standardkodierung!)
9. Vom Benutzer gewähltes Zeichen in einem X-array des Formats U, welches nicht nach den oben aufgeführten Parametern prozentkodiert werden soll (zum Beispiel das Euro-Symbolzeichen, das in der Codepage ISO-8859-1 nicht vorhanden ist) oder um zu verhindern, dass ein Zeichen prozentkodiert wird.
10. Benutzer-definierte Prozentkodierung in ein X-array des Formats A, für ein vom Benutzer gewähltes Zeichen in derselben Ausprägung des X-array.

Der Eingabeparameter für eine Zeichenkette hat das Natural-Format U. Das bedeutet, dass die Eingabezeichenkette alle Unicode-Zeichen enthalten darf. Die Ausgabezeichenkette der erweiterten Funktion hat das Natural-Format A und ist in der Natural-Standard-Codepage (\*CODEPAGE) kodiert. Die Codepage der Prozentkodierung kann gewählt werden. Die Prozentkodierung gemäß UTF-8, ISO-8859-1, des Euro-Symbolzeichens erfolgt mittels eines `MOVE ENCODED`-Statements. Ist ein Eingabezeichen in der für die Prozentkodierung verwendeten Ziel-Codepage nicht vorhanden, dann wird dieses Zeichen nicht kodiert. Das bedeutet, das Zeichen wird unverändert in der Natural-Standard-Codepage zurückgegeben. Falls das Zeichen in der Natural-Standard-Codepage auch nicht vorhanden ist, wird es durch das vom `MOVE ENCODED`-Statement zurückgelieferte Ersetzungszeichen ersetzt. Das Ersetzungszeichen wird prozentkodiert. Dieser Fall kann nur dann eintreten, wenn die Codepage zur Prozentkodierung nicht UTF-8 ist. Der zuletzt aufgetretene `MOVE ENCODED`-Fehler wird zurückgeliefert.

Die Parameter sind optional. Wenn der Benutzer einen Parameter nicht angibt, wird der Standardvorgabewert verwendet. Falls der Benutzer eine eigene Zeichenumsetzungstabelle angibt, werden die Zeichen in der Tabelle entsprechend dieser Tabelle und nicht entsprechend den anderen Parametern prozentkodiert. Wenn die Prozentkodierung eines Zeichens in der benutzerdefinierten Umsetzungstabelle gleich dem Zeichen oder leer ist, wird dieses Zeichen nicht kodiert. Somit

können einzelne Zeichen aus dem reservierten oder nicht reservierten Zeichensatz ausgeschlossen werden.

Folgende Funktionen stehen zur Verfügung:

- **SAGENCE** - Erweiterte Kodierung (Format U nach Format A, optionale Parameter)
- **SAGDECE** - Erweiterte Dekodierung (Format A nach Format U, optionale Parameter)

**SAGENCE - Erweiterte Kodierung (Format U nach Format A, optionale Parameter)**

Die Funktion SAGENCE dient zur Prozentkodierung einer Zeichenkette unter Verwendung des Hexadezimalwerts aus der gewählten Codepage (standardmäßi UTF-8). Gemäß dem Standard RFC3986 werden reservierte Zeichen und Zeichen unterhalb von US-ASCII x'7F', welche in einer URL nicht zulässig sind, prozentkodiert. Außerdem werden das Zeichen für den Leerschritt und das Prozentzeichen (%) kodiert.

Darüber hinaus werden nicht reservierte Zeichen gemäß RCF3986 und Zeichen oberhalb von US-ASCII x'7F', zum Beispiel deutsche Umlaute, von dieser Funktion kodiert.

SAGENCE benötigt Natural-Unicode-Unterstützung.

| Objekt  | Beschreibung  |
|---------|---|
| SAGENCE | Dies ist der Aufruf der erweiterten Kodierfunktion.   |
|         | Parameter:  |
|         | <pre> P-DEC-STR-E      (U) P-RET            (I4)      OPTIONAL /* 0:      ok                                /* else: Natural error returned                                /*          by the GIVING clause of                                /*          MOVE ENCODED.                                /*          This is the error which                                /*          comes up when a character                                /*          cannot be converted into                                /*          the target code page.  /* Error strategy: /* Step 1: If a character shall be %-encoded and is not available /* in the code page for %-encoding, the character will not be /* %-encoded. It will be copied. /* Step 2: If a character will not be %-encoded but copied from the /* input format U-variable to a format A-variable (in *CODEPAGE) /* and the character is not available in *CODEPAGE, a substitution /* character will be used instead. The substitution character will /* be %-encoded. /* The last error will be returned in P-RET. P-ERR-CHAR      (U1)      OPTIONAL /* Character causing the error P-SPACE         (A1)      OPTIONAL /* '%' =&gt; %20                                /* ' ' =&gt; ' '                                /* else =&gt; '+' (default) P-UNRES         (A1)      OPTIONAL /* 'E' =&gt; encode                     </pre> |

| Objekt   | Beschreibung   |
|----------|--|
|          | <pre> /* else =&gt; don't encode (default) P-RES          (A1)    OPTIONAL /* 'E' =&gt; encode (default) /* else =&gt; don't encode P-OTHER        (A1)    OPTIONAL /* 'E' =&gt; encode (default) /* else =&gt; don't encode P-CP           (A64)   OPTIONAL /* IANA name e.g. UTF-8 (default) /* or ISO-8859-1 /* On mainframe only code page names defined with the macro NTCPAGE /* in the source module NATCONFIG can be used. Other code page names /* are rejected with a corresponding runtime error. /* P-CP-TABLE-CHAR(U1/1:*) OPTIONAL /* user selected char to be /* %-encoded, e.g. 'ö' or '/' P-CP-TABLE-ENC (A12/1:*) OPTIONAL /* user %-encoding /* e.g. character 'ö' /*      '%F6'   -&gt; ISO-8859-1 /*      '%C3%B6' -&gt; UTF-8 /* e.g. character '/' /*      '/'     -&gt; '/' not encoded /*           although P-RES = 'E' /* Characters in this table will be encoded according to the /* specified %-encoding. If the U12 encoding part is blank (space /* according to *CODEPAGE) or the P-CP-TABLE-ENC value is equal to /* the character, then the character will not be encoded at all. /* </pre> |
| SAGENCEP | <p>Der Copycode, der die Prototyp-Definition enthält, wird nur bei der Kompilierung benutzt, um den Typ der Rückgabewaren als Referenz für den Funktionsaufruf zu bestimmen und um die Parameter zu prüfen, falls dies gewünscht wird.</p> <p>SAGENCEP ist optional.</p>   |
| URLX01   | <p><b>Beispiel-Programm</b> in der Library SYSEXPB.</p> <p>Beispielaufufe</p> <p>Verwendet werden die Standardvorgabewerte:</p> <pre>#URL-ENC := SAGENCE(&lt;&lt;#URL-DEC-U&gt;&gt;)</pre> <p>Alle Parameter, die möglich sind, werden angegeben:</p> <pre>#URL-ENC := SAGENCE(&lt;&lt;#URL-DEC-U, L-RET, L-ERR-CHAR, L-SPACE, L-UNRES, ↵ L-RES, L-OTHER, L-CP, L-CP-TAB-CHAR(*), L-CP-TAB-ENC(*) &gt;&gt;)</pre>  |

### SAGDECE - Erweiterte Dekodierung (Format A nach Format U, optionale Parameter)

Die Funktion SAGDECE dient zum Dekodieren der von der Funktion SAGENCE gelieferten Prozentkodierungen. Falls ein Leerschrittzeichen und/oder eine Codepage angegeben werden, müssen die Werte mit den bei der Kodierung angegebenen Werte übereinstimmen.

SAGDECE benötigt Natural-Unicode-Unterstützung.

| Objekt   | Beschreibung  |
|----------|---|
| SAGDECE  | <p>Dies ist der Aufruf der erweiterten Dekodierfunktion.</p> <p>Parameter:</p> <pre> 1 P-ENC-STR-E      (A) 1 P-RET            (I4)    OPTIONAL /* 0:      ok                                   /* else: Natural error returned                                   /*         by the GIVING clause of                                   /*         MOVE ENCODED.                                   /*         This error comes up                                   /*         when a %-encoded                                   /*         character cannot be                                   /*         converted into the                                   /*         target code page.  /* The last error will be returned in P-RET. 1 P-ERR-CHAR      (A12)    OPTIONAL /* Error character %-encoded 1 P-SPACE         (A1)     OPTIONAL /* ' ' =&gt; ' '                                   /* else =&gt; '+' (default) 1 P-CP            (A64)    OPTIONAL /* IANA name e.g. UTF-8 (default)                                   /* or ISO-8859-1  /* On mainframe only code page names defined with the macro NTCPAGE /* in the source module NATCONFIG can be used. Other code page names /* are rejected with a corresponding runtime error. /* </pre> |
| SAGDECEP | <p>Der Copycode, der die Prototyp-Definition enthält, wird nur bei der Kompilierung benutzt, um den Typ der Rückgabewerte als Referenz für den Funktionsaufruf zu bestimmen und um die Parameter zu prüfen, falls dies gewünscht wird.</p> <p>SAGDECEP ist optional.</p>  |
| URLX01   | <p><b>Example program</b> in der Library SYSEXPG.</p> <p>Beispielaufufe</p> <p>Verwendet werden die Standardvorgabewerte:</p>   |

| Objekt | Beschreibung   |
|--------|--|
|        | <pre>#URL-DEC-U := SAGDECE(&lt;#URL-ENC&gt;)</pre> <p>Alle Parameter, die möglich sind, werden angegeben:</p> <pre>#URL-DEC-U := SAGDECE(&lt;#URL-ENC,L-RET,L-ERR-CHAR-DEC,L-SPACE,L-CP&gt;)</pre> |

## Beispiel-Programm

Dieses Beispiel-Programm befindet sich in der Library SYSEXPB:

```
** Example 'URLX01': ENCODED-STR := SAGENC(<DECODED-STR>)
*****
DEFINE DATA
LOCAL
1 SAMPLE-STRING (A72)
/*
1 #URL-DEC      (A) DYNAMIC
1 #URL-ENC      (A) DYNAMIC
/*
1 #URL-DEC-U    (U) DYNAMIC
/*
1 L-RET         (I4)  /* Return code
1 L-ERR-CHAR    (U1)  /* Error character
1 L-ERR-CHAR-DEC(A12) /* Decoded error character
1 L-SPACE       (A1)  /* '%' => %20, ' ' => ' ',
/* else => '+' (default)
1 L-UNRES       (A1)  /* 'E' => encode, else => don't encode (default)
1 L-RES         (A1)  /* 'E' => encode (default), else => don't encode
1 L-OTHER       (A1)  /* 'E' => encode (default), else => don't encode
1 L-CP          (A64) /* default *CODEPAGE
1 L-CP-TAB-CHAR (U1/1:1)
1 L-CP-TAB-ENC  (A12/1:1)
1 L-MSG         (U72)
END-DEFINE
/*
/*
/*
WRITE 'Sample string to be processed:'
/* The string below shall be encoded and decoded again.
/* After decoding it should be unchanged.
SAMPLE-STRING := '"Decoded data!'"
WRITE SAMPLE-STRING (AL=72) /
/*
/* Assign the sample string to the input variable #URL-DEC of the
/* simple encoding function.
#URL-DEC      := SAMPLE-STRING
/*
/* Copycode SAGENC containing the prototype definition is used at
```

```

/* compilation time only in order to determine the type of the return
/* variable for function call reference and to check the parameters,
/* if this is desired. SAGENC is optional.
INCLUDE SAGENC
/*
/* SAGENC(<<#URL-DEC>) is the simple encoding function call.
/*
/* Function SAGENC %-encodes a string to code page ISO-8859-1.
/* According to standard RFC3986 reserved characters and characters
/* below US-ASCII x'7F' which are not allowed in a URL will be
/* %-encoded.
/* Also the space and the percent sign will be encoded.
/* Unreserved characters according to RFC3986 and characters above
/* US-ASCII x'7F' will not be encoded. If you want to encode such
/* characters, use the extended encoding function.
/*
/* ---- Space                ' ' -> '+'
/* ---- Percent sign         '%' -> '%25'
/*
/* Unreserved characters according to RFC3986 (will not be encoded!):
/* ---- Period (fullstop)   '.' -- '%2E'
/* ---- Tilde                '~' -- '%7E'
/* ---- Hyphen              '-' -- '%2D'
/* ---- Underscore character '_' -- '%5F'
/* ---- digits, lower and upper case characters
/* ---- 0123456789abcdefgijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVWXYZ
/*
/* Reserved characters according to RFC3986:
/* ---- Exclamation mark    '!' -> '%21'
/* ---- Number sign         '#' -> '%23'
/* ---- Dollar sign         '$' -> '%24'
/* ---- Ampersand           '&' -> '%26'
/* ---- Apostrophe         ''' -> '%27'
/* ---- Left parenthesis   '(' -> '%28'
/* ---- Right parenthesis  ')' -> '%29'
/* ---- Asterisk            '*' -> '%2A'
/* ---- Plus sign          '+' -> '%2B'
/* ---- Comma              ',' -> '%2C'
/* ---- Reverse solidus (backslash) '/' -> '%2F'
/* ---- Colon              ':' -> '%3A'
/* ---- Semi-colon         ';' -> '%3B'
/* ---- Equals sign        '=' -> '%3D'
/* ---- Question mark      '?' -> '%3F'
/* ---- Commercial at     '@' -> '%40'
/* ---- Square bracket open '[' -> '%5B'
/* ---- Square bracket close ']' -> '%5D'
/*
/* Other characters below x'7F' (US-ASCII) but not allowed in URL
/* ---- Quotation mark     '"' -> '%22'
/* ---- Less than          '<' -> '%3C'
/* ---- Greater than       '>' -> '%3E'
/* ---- Reverse solidus (backslash) '\ ' -> '%5C'

```

```

/* ---- Accent, Circumflex      '^' -> '%5E'
/* ---- Accent, Grave           '`' -> '%60'
/* ---- Opening brace           '{' -> '%7B'
/* ---- Vertical bar            '|' -> '%7C'
/* ---- Closing brace           '}' -> '%7D'
/*
#URL-ENC := SAGENC(<#URL-DEC>)
/*
/*
WRITE 'Simple function, encoded:'
WRITE #URL-ENC (AL=72)
/*
/* Copycode SAGDECP containing the prototype definition is used at
/* compilation time only in order to determine the type of the return
/* variable for function call reference and to check the parameters,
/* if this is desired. SAGDECP is optional.
INCLUDE SAGDECP
/*
/* SAGDEC(<#URL-ENC>) is the simple decoding function call.
/* It decodes the above described %-encodings.
/*
#URL-DEC := SAGDEC(<#URL-ENC>)
/*
/*
/* The result after encoding and decoding must be equal to the original
/* SAMPLE-STRING.
WRITE 'Simple function, decoded:'
WRITE #URL-DEC (AL=72)
/*
/*
/*
WRITE /
/*
/*
/* Assign the sample string to the input variable #URL-DEC-U of the
/* enhanced encoding function.
#URL-DEC-U := SAMPLE-STRING
/*
/* Copycode SAGENCEP containing the prototype definition is used at
/* compilation time only in order to determine the type of the return
/* variable for function call reference and to check the parameters,
/* if this is desired. SAGENCEP is optional.
INCLUDE SAGENCEP
/*
/* This is the enhanced encoding function call.
/* The way, characters will be %-encoded depends on the input
/* parameter of the function.
/* The parameters of the encoding and decoding function are preset
/* with the default values.
/* L-CP-TAB-CHAR(*) and L-CP-TAB-ENC(*) don't have default values.
/* L-CP-TAB-CHAR(1) = 'ä' and L-CP-TAB-ENC(1) = '%C3%A4' will not be

```

```

/* used for the sample string '"Decoded data! "'. The string does not
/* contain an 'ä'.
L-SPACE      := '+'          /* encoding and decoding
L-UNRES      := 'D'          /* encoding only
L-RES        := 'E'          /* encoding only
L-OTHER      := 'E'          /* encoding only
L-CP         := 'UTF-8'      /* encoding and decoding
                                   /* e.g. ISO-8859-1, UTF-16BE, UTF-32BE
L-CP-TAB-CHAR(1) := 'ä'      /* encoding only
L-CP-TAB-ENC (1) := '%C3%A4' /* encoding only
/*
/* Note that all possible parameters are specified for this sample
/* call.
/* If the default values shall be used and no return code is wanted,
/* all parameters can be omitted, besides the string #URL-DEC-U.
/*
#URL-ENC := SAGENCE(<#URL-DEC-U,L-RET,L-ERR-CHAR,L-SPACE,L-UNRES,
  L-RES,L-OTHER,L-CP,L-CP-TAB-CHAR(*),L-CP-TAB-ENC(*) >)
WRITE 'Extended function, encoded:'
WRITE #URL-ENC (AL=72)
IF L-RET NE 0 THEN
  /* If L-RET = 0, the function worked ok. Else L-RET contains the
  /* Natural error returned by the GIVING clause of MOVE ENCODED.
  /* The error comes up when a character cannot be converted into
  /* the target codepage, e.g. because a character does not exist
  /* in the target codepage.
  COMPRESS 'Error' L-RET 'with MOVE ENCODED of' L-ERR-CHAR INTO L-MSG
  WRITE L-MSG
END-IF
/*
/* Copycode SAGDECEP containing the prototype definition is used at
/* compilation time only in order to determine the type of the return
/* variable for function call reference and to check the parameters,
/* if this is desired. SAGDECEP is optional.
INCLUDE SAGDECEP
/*
/* This is the 1st enhanced decoding function call with 5 parameters.
/* Note that all possible parameters are specified for this sample
/* call.
/* Since the parameters have the default values, the subsequent
/* function calls return the same result although parameters
/* have been omitted.
#URL-DEC-U := SAGDECE(<#URL-ENC,L-RET,L-ERR-CHAR-DEC,L-SPACE,L-CP>)
WRITE 'Extended function, decoded:'
WRITE #URL-DEC-U (AL=72)
IF L-RET NE 0 THEN
  /* If L-RET = 0, the function worked ok. Else L-RET contains the
  /* Natural error returned by the GIVING clause of MOVE ENCODED.
  /* The error comes up when a %-encoded character cannot be converted
  /* into the target codepage, e.g. because a character does not exist
  /* in the target codepage.
  COMPRESS 'Error' L-RET 'with MOVE ENCODED of' L-ERR-CHAR INTO L-MSG

```



```

WRITE L-MSG
RESET L-RET
END-IF
/*
/* This is the 2nd enhanced decoding function call with one parameter.
#URL-DEC-U := SAGDECE(<#URL-ENC>)
WRITE #URL-DEC-U (AL=72)
/* L-RET will not be returned
/*
/* This is the 3rd enhanced decoding function call with 3 parameters.
#URL-DEC-U := SAGDECE(<#URL-ENC,L-RET,2X,L-CP>)
WRITE #URL-DEC-U (AL=72)
IF L-RET NE 0 THEN
  COMPRESS 'Error' L-RET 'with MOVE ENCODED of' L-ERR-CHAR INTO L-MSG
  WRITE L-MSG
  RESET L-RET
END-IF
/*
END

```

## Base64-Kodierung

Dieser Abschnitt beschreibt Natural-Funktionen, die Sie benutzen können, um mittels Base64-Konvertierung Binärdaten in druckbare, netzkompatible Daten bzw. in umgekehrter Richtung zu konvertieren.

Base64-Konvertierung bedeutet eine Umwandlung vom Format B in das Format A und zurück nach Format B. Dabei werden 6 (binäre) Bits in 8 (alphanumerische) Bits umgewandelt. Beispiel: Ein Wert mit Format/Länge B3 wird in einen Wert mit Format/Länge A4 umgewandelt.



**Anmerkung:** Jeder Binärwert wird in einen eindeutigen alphanumerischen Wert umgewandelt. Die Rückumwandlung dieses alphanumerischen Werts ergibt dann wieder den ursprünglichen Binärwert. Dies ist jedoch bei den meisten Umwandlungen von Format A nach Format B und zurück nach Format A nicht der Fall.

Die Konvertierung kann benutzt werden, um eine Datei vom Typ `.bmp` per TCP/IP zu übertragen oder um Natural-Binär- oder Ganzzahlwerte per Utility-Protokoll zu übertragen.

**Nur bei Open Systems:** Es stehen drei Modi zur Verfügung: RFC3548, RFC2045 und NATRPC (Standardeinstellung). NATRPC bedeutet, dass die Umwandlung entsprechend der NATRPC-Ligik erfolgt. Diese ist zu 100% großrechnerkompatibel. RFC2045 ist der Standardvorgabewert für den CMBASE64-Aufruf. RFC3548 ist wie NATRPC, aber nicht benötigte alphanumerische Bytes werden mit einem Gleichheitszeichen (=) gefüllt.

Folgende Funktionen stehen zur Verfügung:

- [SAG64BA](#) - Konvertierung von binär nach alphanumerisch

■ SAG64AB - Konvertierung von alphanumerisch nach binär

Diese beiden Funktionen zusammengenommen bieten die gleiche Funktionalität wie das API USR4210N. Dieses API befindet sich in der Library SYSEXT.

**SAG64BA - Konvertierung von binär nach alphanumerisch**

Die Funktion SAG64BA wandelt mittels Base64-Konvertierung Binärdaten in druckbare, netzkompatible Daten

| Objekt  | Beschreibung  |
|---------|---|
| SAG64BA | Dies ist die Funktion zur Wandlung vom Binärformat in alphanumerisches Format.  |
|         | Parameter:  |
|         | <pre> 1 PARM-B          (B)          DYNAMIC BY VALUE                   /* Binary source input/target output 1 PARM-RC          (I4)         OPTIONAL                   /* 0:      ok                   /* Mainframe                   /* 1 Source is not numeric                   /* 2 Source is not packed                   /* 3 Source is not floating point                   /* 4 Overflow, source doesn't fit into target                   /* 5 Integer overflow                   /* 6 Source is not a valid date or time                   /* 7 Length error (hex input not even)                   /* 8 Target precision is less than source precision                   /* 9 Float underflow (result-&gt;0)                   /* 10 Alpha source contains non-hex characters                   /* 20 Invalid function code                   /* Open Systems                   /* 1 Invalid value for RFC parameter                   /* 2 Invalid function code                   /* 3 CMBASE64: Overflow, source doesn't fit into                   /*           target                   /* 4 CMBASE64: Non-base64 character found in encoded                   /*           data                   /* 5 CMBASE64: Out of memory                   /* 6 CMBASE64: Invalid number of parameters                   /* 7 CMBASE64: Invalid parameter type                   /* 8 CMBASE64: Invalid parameter length                   /* 9 CMBASE64: Invalid function code                   /* 10 CMBASE64: Unkown return code 1 PARM-ERRTXT      (A72)        OPTIONAL                   /* blank, if ok no error                   /* else error text 1 PARM-RFC          (B1)         OPTIONAL                   /* OS only, not used for MF                   /* 0 - RFC3548; 3 - RFC2045; 4 - NATRPC;                 </pre> |

| Objekt   | Beschreibung  |
|----------|---|
| SAG64BAP | <p>Der Copycode, der die Prototyp-Definition enthält, wird nur bei der Kompilierung benutzt, um den Typ der Rückgabewerte als Referenz für den Funktionsaufruf zu bestimmen und um die Parameter zu prüfen, falls dies gewünscht wird.</p> <p>SAG64BAP ist optional.</p>  |
| B64X01   | <p><b>Beispiel-Programm</b> in der Library SYSEXPB.</p> <p>Verwendet werden die Standardvorgabewerte:</p> <pre>PARM-A := SAG64BA(&lt;PARM-B&gt;)</pre> <p>Alle Parameter, die möglich sind, werden angegeben (PARM-RFC gilt nicht bei Großrechnern):</p> <pre>PARM-A := SAG64BA(&lt;PARM-B, PARM-RC, PARM-ERRTXT, PARM-RFC&gt;)</pre> |

### SAG64AB - Wandlung vom alphanumerischen Format in das Binärformat

Die Funktion SAG64AB wandelt mittels Base64-Konvertierung druckbare, netzkompatible Daten in Binärdaten.

| Objekt  | Beschreibung  |
|---------|---|
| SAG64AB | <p>Dies ist die Funktion zur Wandlung vom alphanumerischen Format in das Binärformat.</p> <p>Parameter:</p> <pre>1 PARM-A      (A)               /* Alpha source input/target output 1 PARM-RC    (I4)  OPTIONAL               /* 0:   ok               /* Mainframe               /* 1 Source is not numeric               /* 2 Source is not packed               /* 3 Source is not floating point               /* 4 Overflow, source doesn't fit into target               /* 5 Integer overflow               /* 6 Source is not a valid date or time               /* 7 Length error (hex input not even)               /* 8 Target precision is less than source precision               /* 9 Float underflow (result-&gt;0)               /* 10 Alpha source contains non-hex characters               /* 20 Invalid function code               /* Open Systems               /* 1 Invalid value for RFC parameter               /* 2 Invalid function code               /* 3 CMBASE64: Overflow, source doesn't fit into               /*      target               /* 4 CMBASE64: Non-base64 character found in encoded               /*      data</pre> |

| Objekt   | Beschreibung   |
|----------|--|
|          | <pre> /* 5 CMBASE64: Out of memory /* 6 CMBASE64: Invalid number of parameters /* 7 CMBASE64: Invalid parameter type /* 8 CMBASE64: Invalid parameter length /* 9 CMBASE64: Invalid function code /* 10 CMBASE64: Unkown return code 1 PARM-ERRTXT (A72) OPTIONAL /* blank, if ok no error /* else error text 1 PARM-RFC (B1) OPTIONAL /* OS only, not used for MF /* 0 - RFC3548; 3 - RFC2045; 4 - NATRPC;                     </pre> |
| SAG64ABP | <p>Der Copycode, der die Prototyp-Definition enthält, wird nur bei der Kompilierung benutzt, um den Typ der Rückgabewaren als Referenz für den Funktionsaufruf zu bestimmen und um die Parameter zu prüfen, falls dies gewünscht wird.</p> <p>SAG64ABP ist optional.</p>   |
| B64X01   | <p><b>Beispiel-Programm</b> in der Library SYSEXPB.</p> <p>Verwendet werden die Standardvorgabewerte:</p> <pre> PARAM-B := SAG64AB(&lt;PARAM-A&gt;)                     </pre> <p>Alle Parameter, die möglich sind, werden angegeben (PARAM-RFC gilt nicht bei Großrechnern):</p> <pre> PARAM-B := SAG64AB(&lt;PARAM-A, PARAM-RC, PARAM-ERRTXT, PARAM-RFC&gt;)                     </pre>  |

## Beispiel-Programm

Beispiel-Programm B64X01 in der Library SYSEXPB:

```

** Example 'B64X01': BASE64-A-STR := SAG64BA(<BASE64-B-STR>)
*****
* Function ..... Convert binary data into printable,
*                   network-compatible data or vice versa using
*                   Base64 encoding.
*
*                   Base64 encoding means (B) -> (A) -> (B),
*                   where 6 (binary) bits will be encoded into 8
*                   (alpha) bits, e.g a (B3) value will be encoded
*                   into a (A4) value.
*
*                   Note: Every binary value will be encoded into
*                   a non-ambiguous alpha value. Re-encoding this
*                   alpha value again will result in the original
*                   binary value. However, this is not the case with
*                   most of the (A) -> (B) -> (A) encodings.
                    
```

```

*
*           The encoding may be used to transfer a .bmp
*           file via TCP/IP, or to transfer Natural binary or
*           integer values via the utility protocol.
*
*           Open Systems only:
*           On Open Systems, there are 3 modes:
*           RFC3548, RFC2045 and NATRPC (default).
*           NATRPC means the encoding follows
*           the NATRPC logic. This is 100% MF compatible.
*           RFC2045 is the default of the CMBASE64 call.
*           RFC3548 is like NATRPC, but alpha bytes not
*           needed are filled with '='.
*
DEFINE DATA
LOCAL
1 FUNCTION          (A2)
/* 'AB' Alpha to binary encoding
/* 'BA' Binary to alpha encoding
1 PARM-RC          (I4)
/* 0:      ok
/* Mainframe
/* 1  Source is not numeric
/* 2  Source is not packed
/* 3  Source is not floating point
/* 4  Overflow, source doesn't fit into target
/* 5  Integer overflow
/* 6  Source is not a valid date or time
/* 7  Length error (hex input not even)
/* 8  Target precision is less than source precision
/* 9  Float underflow (result->0)
/* 10 Alpha source contains non-hex characters
/* 20 Invalid function code
/* Open Systems
/* 1  Invalid value for RFC parameter
/* 2  Invalid function code
/* 3  CMBASE64: Overflow, source doesn't fit into
/*      target
/* 4  CMBASE64: Non-base64 character found in encoded
/*      data
/* 5  CMBASE64: Out of memory
/* 6  CMBASE64: Inalid number of parameters
/* 7  CMBASE64: Invalid parameter type
/* 8  CMBASE64: Invalid parameter length
/* 9  CMBASE64: Invalid function code
/* 10 CMBASE64: Unkown return code
1 PARM-ERRTXT      (A72)
/* blank, if ok no error
/* else error text
1 PARM-A           (A)   DYNAMIC
/* Alpha source input/target output
1 PARM-B           (B)   DYNAMIC

```

```

*          /* Binary source input/target output
1 PARM-RFC      (B1)
                /* OS only, not used for MF
                /* 0 - RFC3548; 3 - RFC2045; 4 - NATRPC;
/*
1 #BACKUP-A     (A) DYNAMIC
1 #BACKUP-B     (B) DYNAMIC
END-DEFINE
/*
/*
SET KEY ALL
/*
/* Copycode SAG64BAP and SAG64ABP containing the prototype definition
/* is used at compilation time only in order to determine the type of
/* the return variable for function call reference and to check the
/* parameters, if this is desired. SAG64BAP and SAG64ABP are optional.
INCLUDE SAG64BAP
INCLUDE SAG64ABP
/*
REPEAT
  RESET PARM-A PARM-B
  REDUCE DYNAMIC PARM-A TO 0
  REDUCE DYNAMIC PARM-B TO 0
  FUNCTION := 'BA'
  PARM-B := H'0123456789ABCDEF'
  INPUT (AD=MIL IP=OFF CD=NE) WITH TEXT PARM-ERRTXT
    // 10T 'Base64 Encoding:' (YEI)
    / 10T '-' (19) (YEI) /
    / 10T 'Function (BA,AB) ..' (TU) FUNCTION (AD=T)
    / 10T 'Alpha In/Output ...' (TU) PARM-A (AL=30)
    / 10T 'Binary In/Output ..' (TU) PARM-B (EM=HHHHHHHH)
    / 10T 'Response .....' (TU) PARM-RC (AD=OD CD=TU)
    / PARM-ERRTXT (AD=OD CD=TU)
  RESET PARM-ERRTXT
  IF *PF-KEY NE 'ENTR'
    ESCAPE BOTTOM
  END-IF
  /*
  RESET #BACKUP-A #BACKUP-B
  REDUCE DYNAMIC #BACKUP-A TO 0
  REDUCE DYNAMIC #BACKUP-B TO 0
  #BACKUP-A := PARM-A
  #BACKUP-B := PARM-B
  /*
  IF FUNCTION = 'BA'
    /* Parameter PARM-RC, PARM-ERRTXT and PARM-RFC are optional
    /* Parameter PARM-RFC does not apply to mainframe
    /* PARM-A := SAG64BA(<PARM-B,PARM-RC,PARM-ERRTXT,PARM-RFC>)
    PARM-A := SAG64BA(<PARM-B,PARM-RC,PARM-ERRTXT>)
    /* PARM-A := SAG64BA(<PARM-B,PARM-RC>)
    /* PARM-A := SAG64BA(<PARM-B>)
  ELSE

```

```

/* Parameter PARM-RC, PARM-ERRTXT and PARM-RFC are optional
/* Parameter PARM-RFC does not apply to mainframe
/* PARM-B := SAG64AB(<PARM-A,PARM-RC,PARM-ERRTXT,PARM-RFC>)
PARM-B := SAG64AB(<PARM-A,PARM-RC,PARM-ERRTXT>)
/* PARM-B := SAG64AB(<PARM-A,PARM-RC>)
/* PARM-B := SAG64AB(<PARM-A>)
END-IF
/*
IF PARM-RC NE 0 THEN
  WRITE 'Encoding' FUNCTION
  WRITE NOTITLE PARM-ERRTXT
ELSE
  IF FUNCTION = 'BA' THEN
    WRITE 'Binary -> Alpha'
    WRITE '=' PARM-B (EM=HHHHHHHHHHHHHHHHHHHHHHHHHHHHHH)
    / '=' PARM-A (AL=50)
    RESET PARM-B
    REDUCE DYNAMIC PARM-B TO 0
    FUNCTION := 'AB'
  ELSE
    WRITE 'Alpha -> Binary'
    WRITE '=' PARM-A (AL=50) /
    '=' PARM-B (EM=HHHHHHHHHHHHHHHHHHHHHHHHHHHHHH)
    RESET PARM-A
    REDUCE DYNAMIC PARM-A TO 0
    FUNCTION := 'BA'
  END-IF
/*
IF FUNCTION = 'BA'
  /* Parameter PARM-RC, PARM-ERRTXT and PARM-RFC are optional
  /* Parameter PARM-RFC does not apply to mainframe
  /* PARM-A := SAG64BA(<PARM-B,PARM-RC,PARM-ERRTXT,PARM-RFC>)
  PARM-A := SAG64BA(<PARM-B,PARM-RC,PARM-ERRTXT>)
  /* PARM-A := SAG64BA(<PARM-B,PARM-RC>)
  /* PARM-A := SAG64BA(<PARM-B>)
ELSE
  /* Parameter PARM-RC, PARM-ERRTXT and PARM-RFC are optional
  /* Parameter PARM-RFC does not apply to mainframe
  /* PARM-B := SAG64AB(<PARM-A,PARM-RC,PARM-ERRTXT,PARM-RFC>)
  PARM-B := SAG64AB(<PARM-A,PARM-RC,PARM-ERRTXT>)
  /* PARM-B := SAG64AB(<PARM-A,PARM-RC>)
  /* PARM-B := SAG64AB(<PARM-A>)
END-IF
IF PARM-RC NE 0 THEN
  WRITE 'Encoding' FUNCTION
  WRITE NOTITLE PARM-ERRTXT
ELSE
  IF FUNCTION = 'BA' THEN
    WRITE 'Binary -> Alpha'
    WRITE '=' PARM-B (EM=HHHHHHHHHHHHHHHHHHHHHHHHHHHHHH)
    / '=' PARM-A (AL=50)
    IF PARM-A = #BACKUP-A THEN

```





# Stichwortverzeichnis

---

## Symbole

- \*MAXVAL  
Systemfunktion, 25
- \*MINVAL  
Systemfunktion, 25
- \*TRANSLATE  
Systemfunktion, 35
- \*TRIM  
Systemfunktion, 39

## A

- ABS  
Systemfunktion, 19
- ATN  
Systemfunktion, 19
- AVER  
Systemfunktion, 10

## C

- COS  
Systemfunktion, 19
- COUNT  
Systemfunktion, 10

## E

- EXP  
Systemfunktion, 19

## F

- FRAC  
Systemfunktion, 19

## I

- INT  
Systemfunktion, 19

## L

- LOG  
Systemfunktion, 19

## M

- Mathematische Funktionen, 19
- MAX  
Systemfunktion, 10
- MIN  
Systemfunktion, 11

## N

- NAVER  
Systemfunktion, 11
- NCOUNT  
Systemfunktion, 11
- NMIN  
Systemfunktion, 12

## O

- OLD  
Systemfunktion, 12

## P

- POS  
Systemfunktion, 45

## R

- RET  
Systemfunktion, 47

## S

- SGN  
Systemfunktion, 19
- SIN  
Systemfunktion, 19
- SORTKEY  
Systemfunktion, 49
- SQRT  
Systemfunktion, 19
- SUM  
Systemfunktion, 12
- Systemfunktionen, v

## T

- TAN  
Systemfunktion, 19
- TOTAL  
Systemfunktion, 13

## V

- VAL  
Systemfunktion, 19